МИНИСТЕРСТВО ОБРАЗОВАНИЯ И НАУКИ

РОССИЙСКОЙ ФЕДЕРАЦИИ

Федеральное государственное бюджетное образовательное учреждение высшего профессионального образования
«Чувашский государственный университет имени И.Н.Ульянова»

АЛАТЫРСКИЙ ФИЛИАЛ

Факультет управления и экономики
Кафедра высшей математики и информационных технологий

МЕТОДЫ ПРИБЛИЖЕННЫХ ВЫЧИСЛЕНИЙ

Направление «Экономика»

Методические указания к контрольной работе

Преподаватель: доц. Федоров Р.В.

2012-2013 учебный год

Алатырь 2012

Цель дисциплины: освоение основных идей, методов, особенностей областей применения и методики использования их как готового инструмента практической работы при проектировании и разработке систем массового обслуживания, математической обработке данных экономических и других задач, построении алгоритмов и организации вычислительных процессов на ПК, получение представления об общих методах приближенного решения различных математических задач, позволяющими количественно обосновать принимаемые экономические решения.

Задачи дисциплины:

- изложить основные сведения о классических численных методах решения различных прикладных задач: прямые и итерационные методы решения систем линейных алгебраических уравнений; решение нелинейных алгебраических и трансцендентных уравнений; интерполирование; дифференцирование и интегрирование; решение дифференциальных уравнений;
- сформировать представление об основных терминах и понятиях данной дисциплины.

Место дисциплины в структуре ООП: дисциплина по выбору студента математического и естественнонаучного цикла (Б.2.ДВ.4.1).

Требования к результатам освоения дисциплины:

процесс изучения дисциплины направлен на формирование следующих компетенций: ОК-1, ОК-4, ОК-6, ОК-13, ПК-4, ПК-5, ПК-10, ПК-13.

В результате изучения дисциплины студент должен:

знать:

- приемы и навыки вычислительных процедур;
- методику выбора оптимального численного метода решения данной задачи,
- приемы математической характеристики точности исходной информации и оценки точности полученного численного решения.

уметь:

- использовать современные компьютерные технологии и пакеты прикладных программ для решения численных задач;
- осуществлять математические постановки простейших экономических задач;
- выбирать методы их решения и интерпретировать получаемые результаты

владеть:

- основными методами решения математических задач и навыками их применения в задачах моделирования экономических процессов.

Содержание дисциплины:

Введение в приближенные методы вычислений. Основные классы задач, решаемых численными методами. Этапы решения задач на ЭВМ. Прямые и итерационные методы решения систем линейных алгебраических уравнений. Метод LU-разложения, метод простых итераций, метод верхней релаксации, метод Зейделя, метод Якоби. Методы решения нелинейных уравнений. Метод простых итераций, метод Ньютона, модифицированный метод Ньютона, метод релаксации, метод секущих. Интерполяционные формулы. Постановка задачи интерполяции. Интерполяционный многочлен Лагранжа; его существование и единственность; оценка погрешности интерполяционной формулы Лагранжа. Разделенные разности. Интерполяционный многочлен Лагранжа в форме Ньютона с разделенными разностями. Численное интегрирование. Простейшие квадратурные формулы: прямоугольников, трапеций; квадратурные формулы Ньютона-Котеса; оценки погрешности этих квадратурных формул. Решение дифференциальных уравнений. Численное дифференцирование, вычислительная погрешность формул численного дифференцирования. Правило Рунге оценки погрешности. Метод Адамса. Метод Рунге-Кутта. Задачи оптимизации. Методы решения задач линейного программирования, методы решения задач нелинейного программирования.

Общие указания

Задание выполняется по вариантам, определяемым по соответствующему алгоритму преподавателем совместно со студентом

Для написания контрольной работы можно использовать один или несколько источников по данной теме. При анализе источников следует обратить внимание на связь материала источника с материалом лекций.

Текст печатается на листах формата А4 через 1,5 интервала. Поля стандартные. Объем не менее 12 страниц. В тексте должны быть ссылки на использованную литературу, список которой приводится в конце. Задания выполняются в строгой последовательности: сначала указывается условие, затем ответ.

Контрольную работу необходимо представить в сроки, указанные в учебном графике. Работы, не отвечающие требованиям методических указаний, не засчитываются.

Контрольная работа оформляется в следующем виде:

1. титульный лист;
2. содержание;
3. затем приводятся:

для теоретических заданий – вариант ответа;

для практических заданий – распечатки результатов выполненной работы на компьютере и описание проделанных действий.

4. список использованной литературы

ЗАДАНИЕ 1. РЕШЕНИЕ НЕЛИНЕЙНОГО УРАВНЕНИЯ С ОДНИМ НЕИЗВЕСТНЫМ.

Исходные теоретические сведения

Уравнение с одним неизвестным можно записать в каноническом виде

$$f(x) = 0$$

Решение уравнения заключается в нахождении корней, т.е. таких значений x, которые обращают уравнение в тождество. В зависимости от того, какие функции входят в уравнение, разделяют два больших класса уравнений - алгебраические и трансцендентные. Функция называется алгебраической, если для получения значения функции по данному значению x нужно выполнить арифметические операции и возведение в степень. К трансцендентным функциям относятся показательная, логарифмическая, тригонометрические прямые и обратные и т.п.

Найти точные значения корней можно лишь в исключительных случаях. Как правило, используются методы приближенного вычисления корней с заданной степенью точности E. Это означает, что если установлено, что искомый корень лежит внутри интервала [a,b], где a - левая граница, а b - правая граница интервала, и длина интервала (b-a) <= E, то за приближенное значение корня можно принять любое число, находящееся внутри этого интервала.

Процесс нахождения приближенных значений корней разбивается на два этапа: 1) отделение корней и 2) уточнение корней до заданной степени точности. Рассмотрим эти этапы подробнее.

1.1 Отделение корней.

Любой корень уравнения считается отделенным на отрезке [a,b], если на этом отрезке исследуемое уравнение не имеет других корней.

Отделить корни - это значит разбить всю область допустимых значений x на отрезки, в каждом из которых содержится только один корень. Эту операцию можно провести двумя способами - графическим и табличным. Если функция f(x) такова, что можно легко построить качественный график ее изменения, то по этому графику достаточно грубо находятся два числа, между которыми лежит одна точка пересечения функции с осью абсцисс. Иногда с целью облегчения построения, целесообразно представить исходное каноническое уравнение в виде $f_1(x) = f_2(x)$, затем построить графики этих функций, причем абсциссы пересечения графиков и служат корнями данного уравнения.

При наличии компьютера наиболее распространен табличный способ отделения корней. Он заключается в табулировании функции f(x) при изменении x от некоторого значения $x_{нач}$ до значения $x_{кон}$ с шагом dx. Задача заключается в том, чтобы найти в этой таблице такие два смежных значения x, для которых функция имеет разные знаки. Предположим, что такие два значения a и b=a+dx найдены, т.е. f(a)*f(b)<0. Тогда согласно теореме Больцано-Коши внутри отрезка [a,b], если функция f(x) непрерывна, существует точка c, в которой f(c)=0. EXCEL позволяет легко реализовать оба способа отделения корней. Рассмотрим их на примере.

Пример 1.1.

Требуется отделить корни уравнения

$$e^x-10x = 0$$

Для этого надо протабулировать функцию f(X) = exp(X) - 10*X, записанную по правилам EXCEL, и построить ее график при изменении X от какого-то $X_{нач}$ до $X_{кон}$ с шагом dX. Пусть эти значения сначала будут таковы: $X_{нач} = 0$, $X_{кон} = 5$, $dX = 0,5$. Если в этих пределах изменения X нам не удастся отделить ни одного корня, тогда надо будет задать новые начальное и конечное значения x и, может быть, изменить шаг.

Для построения таблицы целесообразно воспользоваться специальной подпрограммой ТАБЛИЦА. Для этого на новом рабочем листе в ячейке B1 введем текст: ОТДЕЛЕНИЕ КОРНЕЙ. Затем в ячейку A2 введем текст: x, а в смежную ей ячейку B2 - текст: f(x). Далее оставим ячейку A3 пустой, но в ячейку B3 введем формулу исследуемой функции по правилам EXCEL, а именно

	A	B
1		ОТДЕЛЕНИЕ КОРНЕЙ
2	x	f(x)
3		=EXP(A3)-10*A3
4	0	
5	0,5	
6	1	

=EXP(A3)-10*A3

Затем заполним числовой ряд изменений X в строках A4:A14 от 0 до 5 с шагом 0,5.

Выделим блок ячеек A3:B14. Теперь дадим команду меню *Данные- Таблица*. Результаты табулирования будут помещены в блок ячеек B4:B14. Для того чтобы сделать их более наглядными, нужно отформатировать блок B4:B14 так, чтобы отрицательные числа окрашивались в красный цвет. В этом случае легко найти два смежных значения X, для которых значения функции имеют разные знаки. Их и надо принять за концы интервала отделения корней. В нашем случае таких интервалов, как видно из таблицы два - [0;0,5] и [3,5;4].

Далее следует построить график нашей функции, выделив блок A4:B14 и вызвав *Мастер Диаграмм*. В результате получим на экране диаграмму изменения f(X), из которой видны следующие интервалы отделения корней [0;1] и [3;4].

Если изменять теперь числовые значения x в блоке A4:A14 то значения функции в ячейках B4:B14и график будут изменяться автоматически.

	A	B
1	ОТДЕЛЕНИЕ КОРНЕЙ с помощью подпрограммы ТАБЛИЦА	
2	x	f(x)
3		1
4	0	1,00
5	0,5	-3,35
6	1	-7,28
7	1,5	-10,52
8	2	-12,61
9	2,5	-12,82
10	3	-9,91
11	3,5	-1,88
12	4	14,60
13	4,5	45,02
14	5	98,41
15		
16		
17		



1.2 Уточнение корней: метод итераций.

Для уточнения корня методом итераций должно быть задано:

1) уравнение f(X) = 0, причем f(X) должно быть задано в виде формулы,

2) числа a - левая граница и b - правая граница интервала, внутри которого лежит один

корень,

3) число E - заданная точность получения корня.

Сам метод можно разбить на два этапа:
а) переход от канонического вида записи уравнения f(X)=0 к итерирующему виду X = g(X),
б) вычислительная итерирующая процедура уточнения корня.

Перейти от канонического вида уравнения к итерирующему можно различными способами, важно лишь чтобы при этом выполнялось **достаточное условие сходимости метода: $|g'(X)| < 1$ на [a,b]**, т.е. модуль первой производной итерирующей функции должен быть меньше 1 на интервале [a,b]. Причем чем меньше этот модуль, тем больше скорость сходимости.

Вычислительная процедура метода состоит в следующем. Выбираем начальное приближение, обычно равное $X_0 = (a+b)/2$. Затем вычислим $X_1=g(X_0)$ и $D= X_1- X_0$. Если модуль D <= E, то X_1 является корнем уравнения. В противном случае переходим ко второй итерации: вычисляем $X_2=g(X_1)$ и новое значение $D=X_2- X_1$. Опять проводим проверку на точность и при необходимости продолжаем итерации. Если g(X) выбрано правильно и удовлетворяет достаточному условию сходимости, то эта итерирующая процедура сойдется к корню. Следует отметить, что от знака g'(X) зависит характер сходимости: **при g'(X)>0 сходимость будет монотонной**, т.е. с увеличением итераций D будет приближаться к E монотонно (не меняя знака), в то время как **при g'(X)<0 сходимость будет колебательной**, т.е. D будет приближаться к E по модулю, меняя знак на каждой итерации.

Рассмотрим реализацию метода итераций на EXCEL на примере.

Пример 1.2

Уточним методом итераций значение корней, отделенных в примере 2.1. Итак пусть f(X)= exp(X) - 10*X, для первого корня a=0 и b=0,5. Пусть E=0,00001. Как выбрать итерирующую функцию? Например, так g(X)=0,1*exp(X). На интервале [a,b] $|g'(X)|<1$ и достаточное условие сходимости выполняется. Кроме того, эта производная >1 на интервале [a,b] и характер сходимости будет монотонный.

	A	B	C
20		Метод итераций	
21	x	g(x)	x-g(x)
22	0	=EXP(A22)*0,1	=A22-B22
23	=B22	=EXP(A23)*0,1	=A23-B23
24	=B23	=EXP(A24)*0,1	=A24-B24
25	=B24	=EXP(A25)*0,1	=A25-B25
26	=B25	=EXP(A26)*0,1	=A26-B26
27	=B26	=EXP(A27)*0,1	=A27-B27
28	=B27	=EXP(A28)*0,1	=A28-B28
29	=B28	=EXP(A29)*0,1	=A29-B29
30	=B29	=EXP(A30)*0,1	=A30-B30
31	=B30	=EXP(A31)*0,1	=A31-B31
32	=B31	=EXP(A32)*0,1	=A32-B32

Запрограммируем метод итераций для этого примера на том же рабочем листе, где мы проводили отделение корней. В ячейку A22 внесем число, равное 0. В ячейку B22 запишем формулу =0,1*EXP(A22), а в ячейку C22 формулу =A22- B22. Таким образом 22 строка содержит данные по первой итерации. Чтобы получить в строке 23 данные по второй итерации, скопируем содержимое ячейки B22 в ячейку A23, записав в A23 формулу =B22. Далее надо скопировать формулы ячеек B22 и C22 в ячейки B23 и C23. Для получения данных всех остальных итераций надо выделить ячейки A23,B23,C23 и скопировать их содержимое в блок A24:C32. После этого следует проанализировать изменение D = X - g(X) в столбце C, найти D<0,00001 по модулю и выбрать соответствующее ему значение X из столбца A. Это и есть приближенное значение корня.

	A	B	C
20	Метод итераций		
21	x	g(x)	x-g(x)
22	0	0,1	-0,1
23	0,1	0,110517	-0,010517092
24	0,110517	0,111686	-0,001168452
25	0,111686	0,111816	-0,000130575
26	0,111816	0,111831	-1,46014E-05
27	0,111831	0,111832	-1,6329E-06
28	0,111832	0,111833	-1,82611E-07
29	0,111833	0,111833	-2,04218E-08
30	0,111833	0,111833	-2,28383E-09
31	0,111833	0,111833	-2,55406E-10
32	0,111833	0,111833	-2,85627E-11
33			
34			



Для большей наглядности можно построить диаграмму для метода итераций. Выделяя блок A22:C32 и используя *Мастер диаграмм*, получим три графика изменения X,g(X) и D в зависимости от номера итераций, для чего на *шаге 3 из 5* выберем формат 2, а на *шаге 4 из 5* построения диаграммы нужно отвести ноль столбцов для меток оси X. Теперь хорошо виден монотонный характер сходимости D.

Для уточнения второго корня этого уравнения на интервале [3,5;4], нужно выбрать другую итерирующую функцию, такую чтобы ее первая производная была по модулю меньше единицы. Выберем g(X)= LN(X)+LN(10). В ячейку A22 внесем новое X0=3,75, а в ячейку B22 - новую формулу =LN(A22)+LN(10). Скопируем формулу из B22 в блок B23:B32 и сразу получим новые данные и перестроенную диаграмму. Определим приближенное значение второго корня.

1.3 Уточнение корней: метод Ньютона.

Для уточнения корня методом Ньютона должно быть дано:

1) уравнение f(X) = 0, причем f(X) должно быть задано в виде формулы,

2) числа a - левая граница и b - правая граница интервала, внутри которого лежит один корень,

3) число E - заданная точность получения корня,

4) функция f(X) должна быть дважды дифференцируемой, причем формулы f'(X) и f''(X) должны быть известны.

Метод состоит в итерационных вычислениях последовательности

$$X_{i+1} = X_i - f(X_i)/f'(X_i), \text{ где } i=0,1,2, ...,$$

исходя из начального приближения X_0, принадлежащего интервалу [a,b] и удовлетворяющего условию $f(X_0)*f'(X_0)>0$. **Достаточные условия сходимости** метода заключаются в том, что первая и вторая производные исследуемой функции должны сохранять знак на интервале [a,b]. В качестве начального приближения выбирают обычно или a, или b, в зависимости от того, кто из них соответствует формуле выбора X_0.

Метод Ньютона допускает простую геометрическую интерпретацию. Если через точку с координатами $(X_i;f(X_i))$ провести касательную к кривой f(X), то абсцисса точки пересечения этой касательной с осью 0X и есть очередное приближение корня X_{i+1}.

Метод Ньютона можно рассматривать как некоторую модификацию метода итераций, дающую наилучшую итерирующую функцию g(X) на каждом шаге итерации. Проведем следующие преобразования с исходным каноническим уравнением f(X)=0. Умножим левую и правую его части на некоторое число λ, отличное от нуля. Затем прибавим слева и справа по X. Тогда будем иметь

$$X = g(X) = X + \lambda * f(X).$$

Дифференцируя g(X), получим $g'(X) = 1 + \lambda * f'(X)$. Из достаточного условия сходимости метода итераций $|g'(X)|<1$. Потребуем, чтобы на i-том шаге итерации сходимость была самой быстрой, т.е. $|g'(X_i)| = 0$. Тогда $\lambda=-1/f'(X_i)$ и мы пришли к методу Ньютона.

Вычислительная процедура метода состоит в следующем. Выбираем начальное приближение X_0, обычно равное a или b. Затем вычислим $X_1= X_0 - f(X_0)/f'(X_0)$ и $D= X_1- X_0$. Если модуль D <= E, то X_1 является корнем уравнения. В противном случае переходим ко второй итерации: вычисляем X_2 и новое значение $D=X_2- X_1$. Опять проводим проверку на точность и при необходимости продолжаем итерации. Если X_0 выбрано правильно, а функция удовлетворяет достаточному условию сходимости, то эта итерирующая процедура быстро сойдется к корню.

Пример 1.3.

Уточним методом Ньютона значение корня, отделенного в примере 1.1. Итак пусть f(X)= exp(X) - 10*X, для первого корня a=0 и b=0,5. Пусть E=0,00001. Формулы для первой и второй производной f(X) таковы

$$f'(X) = \exp(X) - 10 \text{ и } f''(X) = \exp(X).$$

Очевидно, что $X_0 = a = 0$, т.к. $f(0)*f''(0) = 1 > 0$.

Запрограммируем метод Ньютона для этого примера на том же рабочем листе, где мы проводили отделение корней. В ячейку A42 внесем число, равное X_0=0. В ячейку B42 запишем формулу =EXP(A42)-10*A42, в ячейку C42 формулу =EXP(A42)-10, а в ячейку D42 формулу =A42- B42/C42. Затем в ячейку E42 запишем формулу =A42-D42. Таким образом 42 строка содержит данные по первой итерации.

	A	B	C	D	E
40		Метод Ньютона			
41	x	f(x)	f'(x)	x нов	D=x-x нов
42	0	=EXP(A42)-10*A42	=EXP(A42)-10	=A42-B42/C42	=A42-D42
43	=D42	=EXP(A43)-10*A43	=EXP(A43)-10	=A43-B43/C43	=A43-D43
44	=D43	=EXP(A44)-10*A44	=EXP(A44)-10	=A44-B44/C44	=A44-D44
45	=D44	=EXP(A45)-10*A45	=EXP(A45)-10	=A45-B45/C45	=A45-D45
46	=D45	=EXP(A46)-10*A46	=EXP(A46)-10	=A46-B46/C46	=A46-D46
47	=D46	=EXP(A47)-10*A47	=EXP(A47)-10	=A47-B47/C47	=A47-D47

	A	B	C	D	E
40	Метод Ньютона				
41	x	f(x)	f'(x)	x нов	D=x-x нов
42	0	1	-9	0,111111	-0,11111
43	0,111111	0,006408	-8,88248	0,111833	-0,00072
44	0,111833	2,91E-07	-8,88167	0,111833	-3,3E-08
45	0,111833	6,66E-16	-8,88167	0,111833	-6,9E-17
46	0,111833	2,22E-16	-8,88167	0,111833	-2,8E-17
47	0,111833	0	-8,88167	0,111833	0

Чтобы получить в строке 43 данные по второй итерации, скопируем содержимое ячейки D42 в ячейку A43, записав в A43 формулу =D42. Далее надо скопировать формулы ячеек B42, C42, D42, E42 в ячейки B43, C43, D43, E43. Для получения данных всех остальных итераций надо выделить ячейки в 43 строке и скопировать их содержимое в блок A44:E47. После этого следует проанализировать изменение D в столбце E, найти D<0,00001 по модулю и выбрать соответствующее ему значение X из столбца A. Это и есть приближенное значение корня. При правильно введенных формулах метод Ньютона сходится за 3 или 4 итерации. Поэтому строить диаграмму для этого метода нет необходимости.

1.4. Уточнение корней: метод бисекции (деления отрезка пополам).

Для уточнения корня методом бисекции должно быть дано:

1) уравнение f(X) = 0, причем f(X) должна быть задана в виде формулы,

2) числа a - левая граница и b - правая граница интервала, внутри которого лежит один корень,

3) число E - заданная точность получения корня.

Напомним, что на концах интервала функция f(X) имеет разные знаки. Вычислительная процедура метода состоит в том, что на каждом шаге итерации на интервале [a,b] выбирают промежуточную точку c так, чтобы она являясь serединой интервала, т.ет c=(a+b)/2. Тогда интервал разделится этой точкой на два равных отрезка [a,c] и [c,b], длины которых равны (b-a)/2. Из двух полученных отрезков выберем тот, на концах которого функция f(X) принимает значения противоположных знаков. Обозначим его снова как [a,b]. На этом заканчивается первая итерация. Далее новый отрезок [a,b] делим снова пополам и проводим вторую и последующие итерации. Процесс деления отрезка пополам производим до тех пор, пока на каком-либо K-том шаге вновь получающийся отрезок не станет меньше или равным величине точности E. Значение шага K легко рассчитать из формулы

$$(b-a)/2^k <=E,$$

где a и b - начальные значения левой и правой границ интервала.

Метод бисекций сходится для любых непрерывных функций, в том числе и недифференцируемых.

Пример 1.4.

Уточним методом бисекции значение корня, отделенного в примере 1.1. Итак пусть f(X)= exp(X) - 10*X, для первого корня a=0 и b=0,5. Пусть E=0,00001.

	A	B	C	D	E	F	G
50	Метод бисекции						
51	a	b	c=(a+b)/2	f(a)	f(c)	f(a)*f(c)	D=b-a
52	0	0,5	=(A52+B52)/2	=EXP(A52)-A52*10	=EXP(C52)-C52*10	=D52*E52	=B52-A52
53	=ЕСЛИ(F52>0;C52;A52)	=ЕСЛИ(F52<0;C52;B52)	=(A53+B53)/2	=EXP(A53)-A53*10	=EXP(C53)-C53*10	=D53*E53	=B53-A53
54	=ЕСЛИ(F53>0;C53;A53)	=ЕСЛИ(F53<0;C53;B53)	=(A54+B54)/2	=EXP(A54)-A54*10	=EXP(C54)-C54*10	=D54*E54	=B54-A54
55	=ЕСЛИ(F54>0;C54;A54)	=ЕСЛИ(F54<0;C54;B54)	=(A55+B55)/2	=EXP(A55)-A55*10	=EXP(C55)-C55*10	=D55*E55	=B55-A55
56	=ЕСЛИ(F55>0;C55;A55)	=ЕСЛИ(F55<0;C55;B55)	=(A56+B56)/2	=EXP(A56)-A56*10	=EXP(C56)-C56*10	=D56*E56	=B56-A56
57	=ЕСЛИ(F56>0;C56;A56)	=ЕСЛИ(F56<0;C56;B56)	=(A57+B57)/2	=EXP(A57)-A57*10	=EXP(C57)-C57*10	=D57*E57	=B57-A57
58	=ЕСЛИ(F57>0;C57;A57)	=ЕСЛИ(F57<0;C57;B57)	=(A58+B58)/2	=EXP(A58)-A58*10	=EXP(C58)-C58*10	=D58*E58	=B58-A58
59	=ЕСЛИ(F58>0;C58;A58)	=ЕСЛИ(F58<0;C58;B58)	=(A59+B59)/2	=EXP(A59)-A59*10	=EXP(C59)-C59*10	=D59*E59	=B59-A59
60	=ЕСЛИ(F59>0;C59;A59)	=ЕСЛИ(F59<0;C59;B59)	=(A60+B60)/2	=EXP(A60)-A60*10	=EXP(C60)-C60*10	=D60*E60	=B60-A60
61	=ЕСЛИ(F60>0;C60;A60)	=ЕСЛИ(F60<0;C60;B60)	=(A61+B61)/2	=EXP(A61)-A61*10	=EXP(C61)-C61*10	=D61*E61	=B61-A61
62	=ЕСЛИ(F61>0;C61;A61)	=ЕСЛИ(F61<0;C61;B61)	=(A62+B62)/2	=EXP(A62)-A62*10	=EXP(C62)-C62*10	=D62*E62	=B62-A62
63	=ЕСЛИ(F62>0;C62;A62)	=ЕСЛИ(F62<0;C62;B62)	=(A63+B63)/2	=EXP(A63)-A63*10	=EXP(C63)-C63*10	=D63*E63	=B63-A63
64	=ЕСЛИ(F63>0;C63;A63)	=ЕСЛИ(F63<0;C63;B63)	=(A64+B64)/2	=EXP(A64)-A64*10	=EXP(C64)-C64*10	=D64*E64	=B64-A64
65	=ЕСЛИ(F64>0;C64;A64)	=ЕСЛИ(F64<0;C64;B64)	=(A65+B65)/2	=EXP(A65)-A65*10	=EXP(C65)-C65*10	=D65*E65	=B65-A65
66	=ЕСЛИ(F65>0;C65;A65)	=ЕСЛИ(F65<0;C65;B65)	=(A66+B66)/2	=EXP(A66)-A66*10	=EXP(C66)-C66*10	=D66*E66	=B66-A66

Запрограммируем метод бисекции для этого примера на том же рабочем листе, где мы проводили отделение корней. В ячейки A52 и B52 надо внести числовые значения a и b,в ячейку C52 - формулу =(A52+B52)/2. Далее в ячейку D52 внесем формулу =EXP(A52)-10*A52, в ячейку E52 - формулу =EXP(C52)-10*C52, в ячейку F52 - формулу =D52*E52, и, наконец, в ячейку G52 запишем формулу =B52- A52. В строке 52 мы сформировали первую итерацию. На второй итерации значения в ячейках A53 и B53 зависят от знака числа в ячейке F52. Если F52>0, то значение A53 равно C52. В противном случае оно должно быть равно A52. В ячейке B53 наоборот: если F52<0, то значение B53 равно C52, иначе B52.

Разрешить это затруднение поможет встроенная функция EXCEL, которая носит название ЕСЛИ. Сделаем текущей ячейку A53. В строке формул, рядом с зеленой галочкой щелкнем на кнопке с изображением *f(x)*. Так вызывается *Мастер Функций*. В появившемся диалоге выберем в поле *Категории Функции* категорию *Логические*, а в поле *Имя Функции* - имя ЕСЛИ. На втором шаге диалога заполним три свободных поля следующим образом: в поле *Логическое_выражение* внесем “F52>0” (разумеется без кавычек!), в поле *Значение_если_истина* внесем C52, а в поле *Значение_если_ложь* - A52. Щелкнем по кнопке *Закончить*. Вот и все.

То же самое надо проделать с ячейкой B53. Только *Логическое выражение* будет “F52<0”, *Значение_если_истина* будет C52, а *Значение_если_ложь* соответственно B52.

Далее надо скопировать формулы в блоке ячеек C52:G52 в блок C53:G53. После этого вторая итерация будет проведена в строке 53. Для получения следующих итераций достаточно скопировать формулы из строки 53 в блоке A53:E53 в блок A54: E68. Затем, как обычно, следует найти с столбце E такую строку, где значение D будет меньше E. Тогда число в столбце C в этой строке и есть приближенное значение корня.

Можно построить диаграмму изменения значений в столбцах A, B и C, начиная с первой и кончая последней итерацией. Для этого нужно выделить блок ячеек A52:C68. За дальнейшими инструкциями обратитесь к примеру 1.2.

	A	B	C
50	Метод бисекции		
51	a	b	c=(a+b)/2
52	0	0,5	0,25
53	0	0,25	0,125
54	0	0,125	0,0625
55	0,0625	0,125	0,09375
56	0,09375	0,125	0,109375
57	0,109375	0,125	0,117188
58	0,109375	0,117188	0,113281
59	0,109375	0,113281	0,111328
60	0,111328	0,113281	0,112305
61	0,111328	0,112305	0,111816
62	0,111816	0,112305	0,112061
63	0,111816	0,112061	0,111938
64	0,111816	0,111938	0,111877
65	0,111816	0,111877	0,111847
66	0,111816	0,111847	0,111832



1.5 Уточнение коней: подпрограмма EXCEL “*Подбор параметра*”.

EXCEL обладает большим набором средств, позволяющих решить те или иные вычислительные задачи. Для решения нелинейного уравнения предусмотрена подпрограмма *Подбор параметра*. Продемонстрируем действие этой подпрограммы на предыдущем примере.

Уточним значение корня, отделенного в примере 1.1. Итак пусть f(X)= exp(X) - 10*X. Найдем корень, лежащий на интервале [0; 0,5]. Оставим пустой ячейку A70. В ячейку B70 запишем формулу =EXP(A70)-10*A70. Выберем команду меню *Сервис- Подбор параметра*. Откроется диалог *Подбор параметра*, в котором в поле *Установить в ячейке* запишем B70, в поле *Значение* занесем 0 (ноль), в поле *Изменяя ячейку* укажем A70. Щелкнем по кнопке ОК и появится новый диалог, в котором будет показан результат выполнения операции. В окне *Состояние подбора решения* будет показано найденное значение. Теперь если щелкнуть на кнопке ОК, в ячейку A70 будет внесено найденное значение корня, а в ячейку B70 - значение функции.



Для того, чтобы найти другой корень, лежащий на интервале [3,5; 4] необходимо изменить начальное приближение, которое в нашей таблице находится в ячейке A70. Запишем в эту ячейку одну из границ интервала, например, 4, и снова выполним процедуру подбора параметра. Содержимое клеток A70 и B70 изменится, теперь в этих клетках появятся координаты большего корня.

Порядок выполнения задания.

1.Изучить численные методы решения нелинейных уравнений.

2.Для заданного варианта (см. таблицу), используя табличный процессор Microsoft Excel:

- отделить корни уравнения;
- уточнить корни методам итераций, Ньютона и бисекции.

3. Оформить отчет о проделанной работе.

Варианты заданий.

1. $\arccos x^2 - x$=0	2. $e^x - \arccos\sqrt{x}$ =0
3. $\ln\ln x - e^x$=0	4. $e^{\frac{1}{x^2}} - \ln x$=0
5. $\ln x - \dfrac{1}{1+x^2}$=0	6. $\ln^2 x - \dfrac{1}{x}$=0
7. $arctg\left(\dfrac{1}{x}\right) - x^2$=0	8. $e^{-x} - x^3$=0
9. $tgx - \dfrac{1}{x}$=0	10. $arctgx - \dfrac{1}{x}$=0
11. $x^4 - 13x^2 + 36 - \dfrac{1}{x}$=0	12. $\ln\dfrac{1+x}{1-x} - \cos x^2$=0
13. $2x^2 - x^4 - 1 - \ln x$=0	14. $shx - x + 1$=0
15. $x^2 - x^3 - \dfrac{1}{4+x^2}$=0	16. $chx - \dfrac{4x^3}{1+x^2}$=0
17. $x^3 - 3x - 2e^{-x}$=0	18. $arctgx - \ln x$=0
19. $2^{x^2} - \dfrac{1}{x}$=0	20. $\dfrac{1}{3+2\cos x} - x^3$=0

ЗАДАНИЕ 2. РЕШЕНИЕ СИСТЕМ ЛИНЕЙНЫХ УРАВНЕНИЙ

Исходные теоретические сведения

В общем виде система линейных алгебраических уравнений записывается так:

$$a_{11}x_1+a_{12}x_2+\dots+a_{1n}x_n = b_1$$
$$a_{21}x_1+a_{22}x_2+\dots+a_{2n}x_n = b_2$$
$$\dots\dots\dots\dots\dots\dots\dots\dots\dots\dots$$
$$a_{n1}x_n+a_{n2}x_2+\dots+a_{nn}x_n = b_n$$

Совокупность коэффициентов этой системы запишем в виде квадратной матрицы **A** из *n* строк и *n* столбцов

$$\begin{matrix} a_{11} & a_{12} & \dots & a_{1n} \\ a_{21} & a_{22} & \dots & a_{2n} \\ \dots & \dots & \dots & \dots \\ a_{n1} & a_{n2} & \dots & a_{nn} \end{matrix}$$

Используя матричное исчисление, исходную систему уравнений можно записать в виде

A*X = B,

где **X** - вектор- столбец неизвестных размерностью *n*, а **B** - вектор- столбец свободных членов, тоже размерностью *n*.

Эта система называется совместной, если она имеет хотя бы одно решение, и определенной, если она имеет одно единственное решение. Если все свободные члены равны нулю, то система носит название однородной.

Необходимым и достаточным условием существования единственного решения системы является условие DET=0, где DET - определитель матрицы **A**. На практике при вычислениях на компьютере не всегда удается получить точное равенство DET нулю. В том случае, когда DET близко к нулю, системы называются плохо обусловленными. При их решении на компьютере малые погрешности в исходных данных могут привести к существенным погрешностям в решении. Условие DET~0 является необходимым для плохой обусловленности системы, но не достаточным. Поэтому при решении системы на ЭВМ требуется оценка погрешности, связанной с ограниченностью разрядной сетки компьютера.

Существуют две величины, характеризующие степень отклонения полученного решения от точного. Пусть **Хк** - истинное решение системы, **Хс** - решение, полученное тем или иным методом на ЭВМ, тогда погрешность решения: **Е = Хк - Хс**. Вторая величина - невязка, равная **R = В - А*Хс**. В практических расчетах контроль точности осуществляется с помощью невязки, хотя это и не совсем корректно.

2.1. Матричный метод.

EXCEL дает возможность решить систему линейных алгебраических уравнений матричным методом, т.е.

$$\mathbf{X = A^{-1}*B.}$$

Таким образом, алгоритм решения системы матричным методом можно представить в виде следующей последовательности вычислительных процедур:

1) получить матрицу $\mathbf{A^{-1}}$, обратную матрице **А**;
2) получить решение системы по формуле $\mathbf{Xc = A^{-1}*B}$**;**
3) вычислить новый вектор свободных членов **Вс = А*Хс**;
4) вычислить невязку **R = В - Вс**;
5) получить решение системы по формуле $\mathbf{dXc = A^{-1}*R}$;
6) сравнить все компоненты вектора **dХс** по модулю с заданной погрешностью Е: если все они меньше Е, то закончить вычисления, иначе повторить вычисления с п.2, где **Хс = Хс + dХс**.

Рассмотрим матричный метод решения системы с помощью EXCEL на примере.

Пример 2.1.

Решить систему уравнений

$$20{,}9x_1+\ 1{,}2x_2+\ 2{,}1x_3+\ 0{,}9x_4=21{,}7$$

$$1{,}2x_1+21{,}2x_2+\ 1{,}5x_3+\ 2{,}5x_4=27{,}46$$

$$2{,}1x_1+\ 1{,}5x_2+19{,}8x_3+\ 1{,}3x_4=28{,}76$$

$$0{,}9x_1+\ 2{,}5x_2+\ 1{,}3x_3+32{,}1x_4=49{,}72$$

EXCEL имеет следующие встроенные функции, реализующие матричные вычисления:

а) МОБР - обращение матрицы,

б) МУМНОЖ - умножение двух матриц,

в) МОПРЕД - вычисление определителя матрицы.

При использовании этих функций важно правильно и компактно расположить на рабочем листе блоки ячеек, соответствующие исходным и рабочим матрицам и вектор-столбцам. Откроем новый рабочий лист, щелкнув на выбранном Вами ярлычке. Отведем под матрицу **А** блок ячеек A3:D6. Для наглядности заключим его в черную рамку. Для этого выделим блок A3:D6, дадим команду меню *Формат- Ячейки* и в открывшемся диалоге выберем вкладку *Рамка*. Откроется новый диалог, в котором щелкнем по полю *Рамка- Контур* и выберем в поле *Рамка- Стиль* самую толстую ширину линии. Подтвердим свое решение, щелкнув на кнопке ОК. Выделим теперь блок A8:D11 под матрицу $\mathbf{A^{-1}}$ и также заключим его в черную рамку, проделав действия, аналогичные блоку матрицы **А**. Далее выделим блоки ячеек под вектор-столбцы (обведя их черной рамкой): блок F8:F11 - под вектор **В**, блок H8:H11 - под вектор **Хс**, получающийся в результате умножения $\mathbf{A^{-1}*B}$, блок H3:H6 - под вектор **Вс**, получающийся в результате умножения **А*Хс**, причем для наглядности выделим дополнительный блок F3:F6, куда скопируем компоненты вектора **Хс** из блока H8:H11. И наконец, занесем в ячейки E4 и E9 знак умножения *, а в ячейки G4 и G9 знак равенства =, затем, выделяя по очереди столбцы E и G, дадим команду меню *Формат- Столбец - Подгон ширины*. Таким образом мы подготовили рабочий лист к решению нашей задачи.

Внесем исходные данные: числа матрицы **А** в ячейки блока A3:D6, а числа вектора свободных членов **В** - в ячейки блока F8:F11.

	A	B	C	D	E	F	G	H
1	Решение системы линейных алгебраических уравнений							
2	матрица А					вектор Хс		вектор Вс
3	20,9	1,2	2,1	0,9		0,8		21,7
4	1,2	21,2	1,5	2,5	*	1	=	27,46
5	2,1	1,5	19,8	1,3		1,2		28,76
6	0,9	2,5	1,3	32,1		1,4		49,72
7	матрица, обратная А					вектор В		вектор Хс
8	0,048513	-0,00228	-0,00491	-0,00098		21,7		0,8
9	-0,00228	0,04794	-0,00316	-0,00354	*	27,46	+	1
10	-0,00491	-0,00316	0,051376	-0,0017		28,76		1,2
11	-0,00098	-0,00354	-0,0017	0,031525		49,72		1,4
12								
13	Определитель матрицы А =			273166,4				
14								

Начнем выполнение алгоритма с обращения матрицы **А**. Для этого выделим блок A8:D11, куда должен быть помещен результат операции. Этот блок окрасится в черный цвет, за исключением ячейки A8. Щелкнем по кнопке f_x на панели *Стандартная*, осуществив вызов *Мастера Функций*. Откроется диалог, в котором из поля *Категория функций* выберем строку *Мат. и тригонометрия*, а из поля *Имя функции* - строку МОБР. Перейдем ко второму шагу диалога, щелкнув по кнопке *Шаг>*. Здесь в поле *Массив* надо набить с клавиатуры A3:D6, что соответствует блоку ячеек, занятому матрицей **А**. Щелкнув на кнопке *Закончить*, можно увидеть, что в блоке A8:D11 заполнена лишь ячейка A8. Для завершения операции обращения EXCEL требует выполнения еще двух действий. Сначала надо сделать активной строку формул, щелкнув по ней (в любом месте строки!) - курсор мыши примет при этом форму I. Проверкой правильности Ваших действий будет появление слева от строки формул четырех кнопок, в том числе с зеленой галочкой. После этого следует нажать на клавиатуре клавишу “Ctrl”, затем не отпуская ее - клавишу “Shift”, и не отпуская и ее - клавишу “Enter”, т.е. в результате должны быть нажаты все три клавиши одновременно! Вот теперь весь блок A8:D11 будет заполнен числами и можно выделить блок H8:H11, чтобы начать операцию умножения $\mathbf{A^{-1}*B}$.

Выделив этот блок, снова вызовите *Мастер функций* и в поле *Имя функции* - выбирайте функцию МУМНОЖ. Щелкнув по кнопке *Шаг>*, перейдем ко второму шагу диалога, где в поле *Массив1* внесем адрес A8:D11, а в поле *Массив2* - адрес F8:F11. Щелкнем по кнопке *Закончить* и обнаружим, что в блоке H8:H11 заполнена лишь ячейка H8. Активизируем строку формул (должна появиться зеленая галочка!) и по методике, описанной выше, нажмем одновременно три клавиши “Ctrl”-”Shift”-”Enter”. Результат умножения появится в блоке H8:H11.

Для проверки точности полученного решения системы, проведем операцию вычисления **Вс=А*Хс**. С этой целью скопируем только числовые значения (а не формулы!) ячеек из блока H8:H11 в ячейки F3:F6. Сделать это надо следующим образом. Выделим блок H8:H11. Дадим команду меню *Правка- Копировать*. Выделим блок F3:F6. Дадим команду меню *Правка- Специальная вставка*. Откроется диалог, в котором в поле *Вставить* следует выбрать режим *Значения*. Подтвердим свое решение, щелкнув по кнопке ОК.

После этой операции заполнены числами блоки A3:D6 и F3:F6. Можно приступить к умножению матрицы **А** на вектор **Хс**. Для этого надо выделить блок H3:H6, вызвать *Мастер Функций* и, действуя так же, как и при вычислении $\mathbf{Xc=A^{-1}*B}$, получить **Вс**. Как видно из таблицы, числовые значения векторов **B** и **Bc** совпадают, что говорит о хорошей точности вычислений, т.е. невязка в нашем примере равна нулю.

Подтвердим хорошую обусловленность матрицы **A** вычислением ее определителя. Для этого сделаем активной ячейку D13. С помощью *Мастера Функций* вызовем функцию МОПРЕД. В поле массив занесем адрес блока A3:D6. Щелкнув по кнопке *Закончить*, получим в ячейке D13 числовое значение определителя матрицы **A**. Как видно, оно значительно больше нуля, что говорит о хорошей обусловленности матрицы.

2.2. Метод приближенных вычислений.

Одним из наиболее распространенных итерационных методов решения систем линейных алгебраических уравнений, отличающийся простотой и легкостью программирования, является метод приближенных вычислений или метод Якоби.

Пусть надо решить систему

$$a_{11}x_1+a_{12}x_2+a_{13}x_3 = b_1$$
$$a_{21}x_1+a_{22}x_2+a_{23}x_3 = b_2$$
$$a_{31}x_1+a_{32}x_2+a_{33}x_3 = b_3$$

Предположим, что диагональные элементы $a_{11,}$ $a_{22,}$ a_{33} отличны от нуля. В противном случае можно переставить уравнения. Выразим переменные из первого, второго и третьего уравнений соответственно. Тогда

$$x_1 = [b_1-(a_{12}x_2+a_{13}x_3)]/ a_{11}$$
$$x_2 = [b_2-(a_{21}x_1+a_{23}x_3)]/ a_{22}$$
$$x_3 = [b_3-(a_{31}x_1+a_{33}x_3)]/ a_{33}$$

Зададим начальные приближения неизвестных

$$x_1 = x_1^{(0)}$$
$$x_2 = x_2^{(0)}$$
$$x_3 = x_3^{(0)}$$

Подставляя их в правую часть преобразованной системы, получим новое первое приближение

$$x_1^{(1)} = [b_1-(a_{12}x_2^{(0)}+a_{13}x_3^{(0)})]/ a_{11}$$
$$x_2^{(1)} = [b_2-(a_{21}x_1^{(0)}+a_{23}x_3^{(0)})]/ a_{22}$$
$$x_3^{(1)} = [b_3-(a_{31}x_1^{(0)}+a_{33}x_3^{(0)})]/ a_{33}$$

На этом заканчивается первая итерация. Далее, используя вычисленные значения $x_1^{(1)}$, $x_2^{(1)}$ и $x_3^{(1)}$, можно провести следующую итерацию, чтобы найти $x_1^{(2)}$,$x_2^{(2)}$ и $x_3^{(2)}$, Итерационный процесс продолжается до тех пор, пока на какой-либо k-той итерации все значения $x_i^{(k)}$ не станут близкими к $x_i^{(k-1)}$. Близость этих значений можно характеризовать максимальной абсолютной величиной их разности D. Тогда при заданной допустимой погрешности E критерий окончания итерационного процесса можно записать так

$$D = \max [ABS(x_i^{(k)} - x_i^{(k-1)})] <=E \qquad \text{для } i=1,2,3.$$

Достаточные условия сходимости итерационного процесса

$$\left|a_{i,i}\right| = \sum_{j=1}^{3}\left|a_{i,j}\right| \qquad \text{где } j\#i\text{ , } i=1,2,3$$

При этом хотя бы для одного уравнения неравенство должно выполняться строго.

Пример 2.2.

Решим систему примера 2.1 методом Якоби.

Дано:

система уравнений

$$20{,}9x_1 + 1{,}2x_2 + 2{,}1x_3 + 0{,}9x_4 = 21{,}7$$
$$1{,}2x_1 + 21{,}2x_2 + 1{,}5x_3 + 2{,}5x_4 = 27{,}46$$
$$2{,}1x_1 + 1{,}5x_2 + 19{,}8x_3 + 1{,}3x_4 = 28{,}76$$
$$0{,}9x_1 + 2{,}5x_2 + 1{,}3x_3 + 32{,}1x_4 = 49{,}72$$

начальные приближения

$$x_1^{(0)} = b_1 = 21{,}7$$
$$x_2^{(0)} = b_2 = 27{,}46$$
$$x_3^{(0)} = b_3 = 28{,}76$$
$$x_4^{(0)} = b_4 = 49{,}72$$

допустимая погрешность

$$E = 1E-5.$$

Очевидно, что достаточные условия сходимости метода выполняются. Откроем новый рабочий лист EXCEL. Внеся в ячейку A1 текст с названием метода, отведем вторую строку для заголовка таблицы

Ячейка	**Текст заголовка**
A2	№ итерации
B2	X1
C2	X2
D2	X3
E2	X4
F2	DX1
G2	DX2
H2	DX3
I2	DX4
J2	D

Следующая третья строка должна содержать информацию о нулевой итерации, т.е. в ячейку A3 занесем ноль, а в ячейки B3, C3, D3 и E3 – начальные приближения, равные значениям свободных членов уравнения.

Четвертая строка будет содержать формулы для вычисления первой итерации

Ячейка	**Формула**
A4	1
B4	=(21,7 – (1,2*C3+2.1*D3+0.9*E3))/20.9
C4	=(27.46-(1.2*B3+1.5*D3+2.5*E3))/21.2
D4	=(28.76-(2.1*B3+1.5*C3+1.3*E3))/19.8
E4	=(49.72-(0.9*B3+2.5*C3+1.3*D3))/32.1
F4	=ABS(B4-B3)
G4	=ABS(C4-C3)
H4	=ABS(D4-D3)
I4	=ABS(E4-E3)
J4	=МАКС(F4:I4)

Для проведения остальных итераций следует скопировать формулы ячеек B4:J4 в нижние строки с 5 по, например, 15. Если числовые значения в столбце J будут меньше E, решение найдено. В противном случае следует продолжить копирование. Результат решения показан на рисунке.

	A	B	C	D	E	F	G	H	I	J
1	**Метод приближенных вычислений (Якоби)**									
2	**№**	**x1**	**x2**	**x3**	**x4**	**dx1**	**dx2**	**dx3**	**dx4**	**D**
3	0	21,7	27,46	28,76	49,72					
4	1	-5,5692	-7,8311	-4,2641	-2,3629	27,269	35,291	33,024	52,083	**52,083**
5	2	2,0181	2,1909	1,9219	2,4876	7,5873	10,022	6,186	4,8505	**10,022**
6	3	0,6123	0,7517	0,6259	1,2439	1,4059	1,4392	1,296	1,2438	**1,4392**
7	4	0,8787	1,0697	0,8599	1,4478	0,2664	0,3179	0,2339	0,204	**0,3179**
8	5	0,8281	1,014	0,8146	1,4061	0,0505	0,0557	0,0453	0,0417	**0,0557**
9	6	0,8377	1,025	0,8231	1,4137	0,0095	0,011	0,0085	0,0076	**0,011**
10	7	0,8358	1,0229	0,8215	1,4123	0,0018	0,002	0,0016	0,0015	**0,002**
11	8	0,8362	1,0233	0,8218	1,4125	0,0003	0,0004	0,0003	0,0003	**0,0004**
12	9	0,8361	1,0232	0,8217	1,4125	6E-05	7E-05	6E-05	5E-05	**7E-05**
13	10	0,8361	1,0232	0,8217	1,4125	1E-05	1E-05	1E-05	1E-05	**1E-05**
14	11	0,8361	1,0232	0,8217	1,4125	2E-06	3E-06	2E-06	2E-06	**3E-06**

2.3. Метод Гаусса – Зайделя.

Этот метод является модификацией метода приближенных вычислений и отличается от него формулами вычислений первого и последующего приближений.

Пусть надо решить систему

$$a_{11}x_1+a_{12}x_2+a_{13}x_3 = b_1$$
$$a_{21}x_1+a_{22}x_2+a_{23}x_3 = b_2$$
$$a_{31}x_1+a_{32}x_2+a_{33}x_3 = b_3$$

Предположим, что диагональные элементы $a_{11,}$ $a_{22,}$ a_{33} отличны от нуля. В противном случае можно переставить уравнения. Выразим переменные из первого, второго и третьего уравнений соответственно. Тогда

$$x_1 = [b_1-(a_{12}x_2+a_{13}x_3)]/ a_{11}$$
$$x_2 = [b_2-(a_{21}x_1+a_{23}x_3)]/ a_{22}$$
$$x_3 = [b_3-(a_{31}x_1+a_{33}x_3)]/ a_{33}$$

Зададим начальные приближения неизвестных

$$x_1 = x_1^{(0)}$$
$$x_2 = x_2^{(0)}$$
$$x_3 = x_3^{(0)}$$

Подставляя их в правую часть преобразованной системы, получим новое первое приближение

$$x_1^{(1)} = [b_1-(a_{12}x_2^{(0)}+a_{13}x_3^{(0)})]/ a_{11}$$
$$x_2^{(1)}= [b_2-(a_{21}x_1^{(1)}+a_{23}x_3^{(0)})]/ a_{22}$$
$$x_3^{(1)}= [b_3-(a_{31}x_1^{(1)}+a_{33}x_3^{(1)})]/ a_{33}$$

На этом заканчивается первая итерация. В отличии от метода Якоби, здесь использовались не только начальные приближения, но и уже вычисленные значения неизвестных на первой итерации. Далее, используя вычисленные значения $x_1^{(1)}$, $x_2^{(1)}$ и $x_3^{(1)}$, можно провести следующую итерацию, чтобы найти $x_1^{(2)}$,$x_2^{(2)}$ и $x_3^{(2)}$, Итерационный процесс продолжается до тех пор, пока на какой-либо k-той итерации все значения $x_i^{(k)}$ не станут близкими к $x_i^{(k-1)}$. Близость этих значений можно характеризовать максимальной абсолютной величиной их разности D. Тогда при заданной допустимой погрешности E критерий окончания итерационного процесса можно записать так

$$D = \max [ABS(x_i^{(k)} - x_i^{(k-1)})] <=E \qquad \text{для } i=1,2,3.$$

Достаточные условия сходимости итерационного процесса

$$\left|a_{i,i}\right| = \sum_{j=1}^{3}\left|a_{i,j}\right| \qquad \text{где } j\#i \text{ , } i=1,2,3$$

При этом хотя бы для одного уравнения неравенство должно выполняться строго.

Пример 2.3.

Решим систему примера 2.1 методом Гаусса-Зайделя. При этом оформление листа EXCEL останется тем же. Изменятся лишь формулы в четвертой и последующих строках.

Ячейка	**Формула**
A4	1
B4	=(21,7 – (1,2*C3+2.1*D3+0.9*E3))/20.9
C4	=(27.46-(1.2*B4+1.5*D3+2.5*E3))/21.2
D4	=(28.76-(2.1*B4+1.5*C4+1.3*E3))/19.8
E4	=(49.72-(0.9*B4+2.5*C4+1.3*D4))/32.1
F4	=ABS(B4-B3)
G4	=ABS(C4-C3)
H4	=ABS(D4-D3)
I4	=ABS(E4-E3)
J4	=МАКС(F4:I4)

Для проведения остальных итераций следует скопировать формулы ячеек B4:J4 в нижние строки с 5 по, например, 15. Если числовые значения в столбце J будут меньше E, решение найдено. В противном случае следует продолжить копирование. Результат решения показан на рисунке.

	A	B	C	D	E	F	G	H	I	J
1	**Метод Гаусса-Зайделя**									
2	**№**	**x1**	**x2**	**x3**	**x4**	**dx1**	**dx2**	**dx3**	**dx4**	**D**
3	0	21,7	27,46	28,76	49,72					
4	1	-5,5692	-6,2876	-0,5128	2,2155	27,269	33,748	29,273	47,504	**47,504**
5	2	1,3554	0,9936	0,7491	1,4032	6,9246	7,2812	1,2619	0,8123	**7,2812**
6	3	0,8455	1,029	0,8212	1,4118	0,5099	0,0354	0,0721	0,0086	**0,5099**
7	4	0,8359	1,0234	0,8218	1,4125	0,0096	0,0056	0,0006	0,0007	**0,0096**
8	5	0,83612	1,02324	0,82173	1,4125	0,00023	0,00014	4E-05	5,8E-06	**0,00023**
9	6	0,83614	1,02325	0,82173	1,4125	1,2E-05	1,5E-06	1,2E-06	4E-07	**1,2E-05**
10	7	0,83614	1,02325	0,82173	1,4125	4,9E-08	1,3E-07	7,6E-09	1,2E-08	**1,3E-07**

Как видно, в данном случае метод Гаусса – Зайделя оказался быстрее метода приближенных вычислений.

Порядок выполнения задания.

1. Изучить основные методы решения системы линейных уравнений.

2.Для заданного варианта (см. таблицу), используя табличный процессор Microsoft Excel решить систему уравнений матричным методом, методом приближенных вычислений и методом Гаусса-Зайделя.

3. Оформить отчет о проделанной работе.

Варианты заданий.

1. $$\begin{cases} x_1 + x_2 + 2x_3 + 3x_4 = 1; \\ 3x_1 - x_2 - x_3 - 2x_4 = -4; \\ 2x_1 + 3x_2 - x_3 - x_4 = -6; \\ x_1 + 2x_2 + 3x_3 - x_4 = -4. \end{cases}$$

2. $$\begin{cases} x_1 + 2x_2 + 3x_3 - 2x_4 = 6; \\ x_1 - x_2 - 2x_3 - 3x_4 = 8; \\ 3x_1 + 2x_2 - x_3 + 2x_4 = 4; \\ 2x_1 - 3x_2 + 2x_3 + x_4 = -8. \end{cases}$$

3. $$\begin{cases} x_1 + 2x_2 + 3x_3 + 4x_4 = 5; \\ 2x_1 + x_2 + 2x_3 + 3x_4 = 1; \\ 3x_1 + 2x_2 + x_3 + 2x_4 = 1; \\ 4x_1 + 3x_2 + 2x_3 + x_4 = -5. \end{cases}$$

4. $$\begin{cases} x_2 - 3x_3 + 4x_4 = -5; \\ x_1 - 2x_3 + 3x_4 = -4; \\ 3x_1 + 2x_2 - 5x_4 = 12; \\ 4x_1 + 3x_2 - 5x_3 = 5. \end{cases}$$

5. $\begin{cases} x_1 + 3x_2 + 5x_3 + 7x_4 = 12; \\ 3x_1 + 5x_2 + 7x_3 + x_4 = 0; \\ 5x_1 + 7x_2 + x_3 + 3x_4 = 4; \\ 7x_1 + x_2 + 3x_3 + 5x_4 = 16. \end{cases}$

6. $\begin{cases} x_1 + 5x_2 + 3x_3 - 4x_4 = 20; \\ 3x_1 + x_2 - 2x_3 = 9; \\ 5x_1 - 7x_2 + 10x_4 = -9; \\ 3x_2 - 5x_3 = 1. \end{cases}$

7. $\begin{cases} 2x_1 + x_2 - 5x_3 + x_4 = 8; \\ x_1 - 3x_2 - 6x_4 = 9; \\ 2x_1 - x_3 + 2x_4 = -5; \\ x_1 + 4x_2 - 7x_3 + 6x_4 = 0. \end{cases}$

8. $\begin{cases} 2x_1 - x_2 + 3x_3 + 2x_4 = 4; \\ 3x_1 + 3x_2 + 3x_3 + 2x_4 = 6; \\ 3x_1 - x_2 - x_3 + 2x_4 = 6; \\ 3x_1 - x_2 + 3x_3 - x_4 = 6. \end{cases}$

9. $\begin{cases} x_1 + 2x_2 - x_3 + x_4 = 8; \\ 2x_1 + x_2 + x_3 + x_4 = 5; \\ x_1 - x_2 + 2x_3 + x_4 = -1; \\ x_1 + x_2 - x_3 + 3x_4 = 10. \end{cases}$

10. $\begin{cases} 4x_1 + x_2 - x_4 = -9; \\ x_1 - 3x_2 + 4x_3 = -7; \\ 3x_2 - 2x_3 + 4x_4 = 12; \\ x_1 + 2x_2 - x_3 - 3x_4 = 0. \end{cases}$

11. $\begin{cases} 2x_1 - x_2 + x_3 - x_4 = 1; \\ 2x_1 - x_2 - 3x_4 = 2; \\ 3x_1 - x_3 + x_4 = -3; \\ 2x_1 + 2x_2 - 2x_3 + 5x_4 = -6. \end{cases}$

12. $\begin{cases} x_1 + x_2 - x_3 - x_4 = 0; \\ x_2 + 2x_3 - x_4 = 2; \\ x_1 - x_2 - x_4 = -1; \\ -x_1 + 3x_2 - 2x_3 = 0. \end{cases}$

13. $\begin{cases} 5x_1 + x_2 - x_4 = -9; \\ 3x_1 - 3x_2 + x_3 + 4x_4 = -7; \\ 3x_1 - 2x_3 + x_4 = -16; \\ x_1 - 4x_2 + x_4 = 0. \end{cases}$

14. $\begin{cases} 2x_1 + x_3 + 4x_4 = 9; \\ x_1 + 2x_2 - x_3 + x_4 = 8; \\ 2x_1 + x_2 + x_3 + x_4 = 5; \\ x_1 - x_2 + 2x_3 + x_4 = -1. \end{cases}$

15. $\begin{cases} x_1 + x_2 + 2x_3 + 3x_4 = 1; \\ 3x_1 - x_2 - x_3 - 2x_4 = -4; \\ 2x_1 + 3x_2 - x_3 - x_4 = -6; \\ x_1 + 2x_2 + 3x_3 - x_4 = -4. \end{cases}$

16. $\begin{cases} x_1 + 5x_2 = 2; \\ 2x_1 - x_2 + 3x_3 + 2x_4 = 4; \\ 3x_1 - x_2 - x_3 + 2x_4 = 6; \\ 3x_1 - x_2 + 3x_3 - x_4 = 6. \end{cases}$

17. $\begin{cases} x_1 - 4x_2 - x_4 = 2; \\ x_1 + x_2 + 2x_3 + 3x_4 = 1; \\ 2x_1 + 3x_2 - x_3 - x_4 = -6; \\ x_1 + 2x_2 + 3x_3 - x_4 = -4. \end{cases}$

18. $\begin{cases} 5x_1 - x_2 + x_3 + 3x_4 = -4; \\ x_1 + 2x_2 + 3x_3 - 2x_4 = 6; \\ 2x_1 - x_2 - 2x_3 - 3x_4 = 8; \\ 3x_1 + 2x_2 - x_3 + 2x_4 = 4. \end{cases}$

19. $\begin{cases} 4x_1 - 2x_2 + x_3 - 4x_4 = 3; \\ 2x_1 - x_2 + x_3 - x_4 = 1; \\ 3x_1 - x_3 + x_4 = -3; \\ 2x_1 + 2x_2 - 2x_3 + 5x_4 = -6. \end{cases}$

20. $\begin{cases} 2x_1 - x_3 - 2x_4 = -1; \\ x_2 + 2x_3 - x_4 = 2; \\ x_1 - x_2 - x_4 = -1; \\ -x_1 + 3x_2 - 2x_3 = 0. \end{cases}$

ЗАДАНИЕ 3. РЕШЕНИЕ СИСТЕМЫ НЕЛИНЕЙНЫХ УРАВНЕНИЙ

Исходные теоретические сведения

Система нелинейных уравнений в общем виде записывается так

$$f_1(x_1, x_2, \ldots x_n) = 0$$

$$f_2(x_1,x_2,...x_n) = 0$$

....

$$f_n(x_1,x_2,...x_n) = 0$$

где f_i - нелинейные алгебраические функции. Для решения таких систем обычно используются итерационные методы. Ниже будут рассмотрены два метода - метод Ньютона и метод итераций. Успех применения этих методов во многом определяется выбором начальных приближений. Они должны быть достаточно близки к истинному значению. В противном случае итерационный процесс может не сойтись. Следует заметить, что в общем случае формализованных процедур выбора начальных приближений нет. Только в случае системы второго порядка можно предложить такую процедуру, сравнительно легко реализующуюся на EXCEL.

3.1. Выбор начальных приближений.

Пусть задана система нелинейных уравнений 2-го порядка

$$f_1(x_1,x_2) = 0$$

$$f_2(x_1,x_2) = 0$$

Зададим несколько значений x_1 в диапазоне от $x_{нач}$ до $x_{кон}$ с шагом dx. Подставим сначала в $f_1(x_1,x_2) = 0$ значение $x_{нач}$. Тогда это уравнение становится одномерным, зависящим только от x_2, и его можно решить подпрограммой EXCEL *Подбор параметра* (см. 2.4). То же самое можно проделать и для всех остальных значений x_1. В результате мы будем иметь набор значений x_1 и x_2, для которых $f_1(x_1,x_2) = 0$. Далее, для того же самого набора значений x_1, используя подпрограмму *Подбор параметра*, найдем значения x_2, для которых $f_2(x_1,x_2) = 0$. Если теперь построить с помощью EXCEL графики изменения x_2 в зависимости от x_1 для двух этих случаев, то на пересечении этих графиков можно приближенно определить значения начальных приближений по x_1 и x_2. Если графики не пересекаются, следует задать новый диапазон изменения x_1 и повторить процедуру сначала.

Пример 3.1.

Пусть надо решить систему

$$x_1^3 + x_2^3 - 6x_1 + 3 = 0$$

$$x_1^3 - x_2^3 - 6x_2 + 2 = 0$$

Подберем начальные приближения. Выберем $x_{нач} = 0$, $x_{кон} = 1$, dx = 0,2. Откроем новый рабочий лист EXCEL и занесем эти значения x_1 в блок A4:A9. Выделим блок B4:B9 под значения x_2 первой серии, для которой $f_1(x_1,x_2) = 0$, и блок C4:C9 - под значения x_2 второй серии, для которой $f_2(x_1,x_2) = 0$. Блок D4:D9 отведем для функции $f_1(x_1,x_2)$, а блок E4:E9 - для функции $f_2(x_1,x_2)$. Сделаем текущей ячейку D4. В нее запишем формулу =A4^3+B4^3-6*A4+3. В ячейку E4 запишем формулу =A4^3-C4^3-6*C4+2. Теперь выделим блок D4:E4 и скопируем эти формулы в блок ячеек D5:E9. Разумеется, адреса ячеек столбцов A и C в них будут автоматически изменены.

Перейдем к заполнению блока B4:B9. Снова сделаем текущей ячейку D4. Дадим команду меню *Сервис- Подбор параметра*. В открывшемся диалоге в поле *Установить в ячейке* должен быть указан адрес ячейки D4 в абсолютных адресах. В поле *Значение* следует занести ноль, а в поле *Изменяя ячейку* - занести адрес ячейки B4 (можно в относительных адресах). Щелкнем по кнопке ОК. Появится новый диалог *Состояние подбора параметра*. Если решение найдено, то, щелкнув по кнопке ОК, получим в ячейке B4 нужное нам числовое значение. Далее эту процедуру надо повторить для всех ячеек блока D4:D9. В результате будет заполнен блок B4:B9.

Аналогичным образом следует заполнить блок C4:C9, используя блок E4:E9.

Если блоки в столбцах B и C заполнены, можно построить диаграмму. Для этого необходимо выделить блок A3:E9. Затем щелкнуть по кнопке *Мастер Диаграмм* на панели Стандартная. Передвигаясь по диалогу с помощью кнопки *Шаг>*, выполнить все 5 шагов построения диаграммы, причем на *Шаге 2 из 5* выбрать тип *XY-точечная*, а на *Шаге 3 из* 5 - *формат 6*. Анализируя построенную диаграмму, можно сделать вывод о том, что в качестве начальных приближений можно выбрать x1 =0,5 и x2 = 0,5.

	A	B	C	D	E
1	Решение систем нелинейных уравнений				
2	Подбор начальных приближений.				
3	x1	$x2_1$	$x2_2$	$f1(x1,x2_1)$	$f2(x1,x2_2)$
4	0	-1,44232	0,327543	-0,000457	-0,000398
5	0,2	-1,21816	0,328811	0,0003667	-0,000415
6	0,4	-0,87276	0,337588	-0,000799	-3,74E-06
7	0,6	0,726501	0,361448	-0,00055	8,84E-05
8	0,8	1,087763	0,40733	-0,000928	0,000436
9	1	1,259876	0,481405	-0,000214	4,29E-06
10					



3.2 Метод Ньютона.

Пусть задана система нелинейных уравнений 2-го порядка

$$f_1(x_1,x_2) = 0$$

$$f_2(x_1,x_2) = 0$$

причем левые части уравнений известны в виде формул. Заданы также числовые значения начальных приближений x_{10} и x_{20}, а также E - точность вычислений значений корней. Функции должны быть дифференцируемы и формулы частных производных тоже должны быть известны.

Исходную систему можно записать в матричном виде

$$\mathbf{F(X)} = 0,$$

где **X** - двумерный вектор- столбец с компонентами { x_1,x_2 }, а **F** - двумерный вектор-функция. Метод Ньютона - это метод последовательных приближений по формуле

$$\mathbf{X_{i+1} = X_i - P_i},$$

где $\mathbf{P_i = J_i^{-1} * F_i}$,

i - номер итерации, (i = 0,1,2,...)

$\mathbf{J_i^{-1}}$ - матрица, обратная матрице J на i-той итерации,

J - матрица Якоби, т.е. матрица первых частных производных:

$$\begin{matrix} df_1/dx_1 & df_1/dx_2 \\ df_2/dx_1 & df_2/dx_2. \end{matrix}$$

Таким образом на каждой итерации вычисляется вектор **P**, его компоненты сравниваются с заданной погрешностью E по формуле

$$D=(p_{1i}^2+p_{2i}^2)^{(1/2)},$$

причем когда D <= E , вычисления прекращаются и вектор $\mathbf{X_i}$ считается решением. В противном случае вычисляются новые значения **X** и выполняется следующая итерация.

Достаточным условием сходимости метода служит неособенность матрицы Якоби, т.е. ее определитель (якобиан) не должен быть равным нулю на любой итерации.

Пример 3.2.

Решим систему из примера 3.1 методом Ньютона. Начальные приближения также возьмем из того же примера. Пусть E = 0,00001. Выпишем формулы частных производных:

$df_1/dx_1 = 3x_1^2 - 6$ $\qquad$ $df_1/dx_2 = 3x_2^2$

$df_2/dx_1 = 3x_1^2$ $\qquad$ $df_2/dx_2 = -3x_2^2 - 6.$

Для решения задачи воспользуемся встроенными в EXCEL матричными функциями и процедурами так, как это сделано в разделе 2 настоящего пособия при решении систем линейных уравнений.

Проведем вычисления на том же рабочем листе, что и в примере 3.1. Необходимо

	A	B	C	D	E	F
12	Метод Ньютона.					
13	вектор X		матрица J			якобиан
14	0,5		=3*A14^2-6	=3*A15^2		=МОПРЕД(C14:D15)
15	0,5		=3*A14^2	=-3*A15^2-6		
16	матрица , обратная J			вектор F		вектор P
17	=МОБР(C14:D15)	=МОБР(C14:D15)		=A14^3+A15^3-6*A14+3		=МУМНОЖ(A17:B18;D17:D18)
18	=МОБР(C14:D15)	=МОБР(C14:D15)		=A14^3-A15^3-6*A15+2		=МУМНОЖ(A17:B18;D17:D18)
19						погрешность D
20						=(F17^2+F18^2)^(1/2)

отвести блоки ячеек для векторов **X,F** и **P**, для матриц **J** и $\mathbf{J}^{-1}$, а также ячейки для вычисления якобиана и текущей величины погрешности D. Затем занести начальные приближения в блок **X** и формулы в блоки **J** и **F**. Далее с помощью *Мастера функций* надо организовать вычисление якобиана функцией МОПРЕД , матрицы $\mathbf{J}^{-1}$ - функцией МОБР и вектора P - функцией МУМНОЖ по аналогии с примером 2.1. В результате будет выполнена первая итерация метода Ньютона и по численному значению D следует принять решение о проведении дальнейших итераций.

	A	B	C	D	E	F
12	Метод Ньютона.					
13	вектор X		матрица J			якобиан
14	0,5		-5,25	0,75		34,875
15	0,5		0,75	-6,75		
16	матрица , обратная J			вектор F		вектор P
17	-0,19355	-0,02151		0,25		-0,02688172
18	-0,02151	-0,15054		-1		0,14516129
19						погрешность D
20						0,147629357

Из таблицы ясно, что D>E и дальнейшие итерации необходимы. По формуле Ньютона для получения новых числовых значений вектора **X** нужно из значений блока **X** вычесть значения блока **P**. Это можно выполнить таким образом. Выделим блок **P** и дадим команду меню *Правка-Копировать*. Затем выделим блок **X** и дадим команду меню *Правка- Специальная вставка*. В появившемся диалоге выберем в поле *Вставить* переключатель *Значения*, а в поле *Операция* - переключатель *Вычесть* и подтвердим свой выбор щелчком по кнопке ОК. В результате будет выполнена вторая итерация. Блок ячеек **P** будет обрамлен бегущей пунктирной линией. Если значение D получится все еще большим чем E, то следует снова выделить блок **X** и повторить команду меню *Правка- Специальная вставка* с указанием тех же переключателей. Эти манипуляции можно проводить до тех пор, пока D не станет меньше, чем E. Во время проведения итераций нужно визуально контролировать числовое значение якобиана для выполнения достаточных условий сходимости метода.

3.3. Метод итераций.

Пусть задана система нелинейных уравнений 2-го порядка

$$f_1(x_1,x_2) = 0$$

$$f_2(x_1,x_2) = 0$$

причем левые части уравнений известны в виде формул. Заданы также числовые значения начальных приближений x_{10} и x_{20}, а также E - точность вычислений значений корней.

Для применения итераций исходная система приводится к виду

$$x_1 = g_1(x_1,x_2)$$

$$x_2 = g_2(x_1,x_2),$$

где функции g_i называются итерирующими. Алгоритм решения задается итерирующими формулами

$$x_{1i+1} = g_1(x_{1i},x_{2i})$$

$$x_{2i+1} = g_2(x_{1i},x_{2i}),$$

где i -номер итерации, i = 0,1,2,... Для прекращения итераций вычисляются значения

$$p_{1i+1} = x_{1i+1} - x_{1i}$$

$$p_{2i+1} = x_{2i+1} - x_{2i},$$

$$D=(p_{1i}^2+p_{2i}^2)^{(1/2)},$$

и D сравнивается с E. Итерации продолжаются до тех пор, пока не выполнится условие D<=E. Чтобы процесс вычислений сходился к этому условию, нужно выполнение **достаточного условия сходимости**:

$$|dg_1/dx_1| + |dg_1/dx_2| < 1,$$

$$|dg_2/dx_1| + |dg_2/dx_2| < 1.$$

Возможно также суммирование по столбцам.

Пример 3.3.

Решим систему из примера 3.1 методом итераций. Начальные приближения также возьмем из того же примера. Пусть E = 0,00001. Выпишем формулы для итерирующих функций

$g_1(x_1,x_2) = (x_1^3 + x_2^3 + 3)/6$

$g_2(x_1,x_2) = (x_1^3 - x_2^3 + 2)/6.$

При изменении независимых переменных в пределах $0<= x_1<=1$ и $0<= x_2<=1$ достаточное условие сходимости выполняется, т.к.

$$|dg_1/dx_1| + |dg_1/dx_2| = (x_1^2)/2 + (x_2^2)/2,$$

$$|dg_2/dx_1| + |dg_2/dx_2| = (x_1^2)/2 + |-(x_2^2)/2| .$$

Проведем вычисления на том же рабочем листе, что и в примере 4.2.

Отведем столбец А, начиная с 26 строки под значения x_1, столбец В под значения x_2, столбец С - под g_1, столбец D - под g_2, следующие три столбца под p_1,p_2 и D.

	A	B	C	D	E	F	G
24			Метод итераций.				
25	x1	x2	g1(x1,x2)	g2(x1,x2)	p1(x1,x2)	p2(x1,x2)	D
26	0,5	0,5	=(A26^3+B26^3+3)/6	=(A26^3-B26^3+2)/6	=A26-C26	=B26-D26	=(E26^2+F26^2)^(1/2)
27	=C26	=D26	=(A27^3+B27^3+3)/6	=(A27^3-B27^3+2)/6	=A27-C27	=B27-D27	=(E27^2+F27^2)^(1/2)
28	=C27	=D27	=(A28^3+B28^3+3)/6	=(A28^3-B28^3+2)/6	=A28-C28	=B28-D28	=(E28^2+F28^2)^(1/2)
29	=C28	=D28	=(A29^3+B29^3+3)/6	=(A29^3-B29^3+2)/6	=A29-C29	=B29-D29	=(E29^2+F29^2)^(1/2)
30	=C29	=D29	=(A30^3+B30^3+3)/6	=(A30^3-B30^3+2)/6	=A30-C30	=B30-D30	=(E30^2+F30^2)^(1/2)
31	=C30	=D30	=(A31^3+B31^3+3)/6	=(A31^3-B31^3+2)/6	=A31-C31	=B31-D31	=(E31^2+F31^2)^(1/2)
32	=C31	=D31	=(A32^3+B32^3+3)/6	=(A32^3-B32^3+2)/6	=A32-C32	=B32-D32	=(E32^2+F32^2)^(1/2)

В строке 27 сформируем формулы для второй итерации, а затем скопируем их в блок A28:G32, с учетом изменений относительных адресов ячеек. В результате будем иметь заполненную таблицу

	A	B	C	D	E	F	G
22	Метод итераций.						
23	x1	x2	g1(x1,x2)	g2(x1,x2)	p1(x1,x2)	p2(x1,x2)	D
24	0,5	0,5	0,541667	0,333333	-0,04167	0,166667	0,171796
25	0,541667	0,333333	0,532661	0,353648	0,009006	-0,02031	0,022222
26	0,532661	0,353648	0,53256	0,35115	0,000101	0,002498	0,0025
27	0,53256	0,35115	0,532391	0,351291	0,000169	-0,00014	0,00022
28	0,532391	0,351291	0,532375	0,351258	1,53E-05	3,27E-05	3,61E-05
29	0,532375	0,351258	0,532371	0,351258	4,19E-06	1,54E-07	4,19E-06
30	0,532371	0,351258	0,532371	0,351258	6,03E-07	5,84E-07	8,4E-07

Как видно, процесс итераций сходится достаточно быстро.

Порядок выполнения задания.

1. Изучить основные методы решения системы нелинейных уравнений.

2.Для заданного варианта (см. таблицу), используя табличный процессор Microsoft Excel выбрать начальные приближения для решения системы нелинейных уравнений и решить методом Ньютона и итераций.

3. Оформить отчет о проделанной работе.

Варианты заданий.

1. $\begin{cases} \sin(x+1) - y = 1{,}2; \\ 2x + \cos y = 2. \end{cases}$	2. $\begin{cases} \sin x - 2y = 1{,}6; \\ \cos(y+0{,}5) + x = 0{,}8. \end{cases}$
3. $\begin{cases} \cos(x-1) + y = 0{,}5; \\ x - \cos y = 3. \end{cases}$	4. $\begin{cases} \sin(y-1) + x = 1{,}3; \\ y - \sin(x+1) = 0{,}8. \end{cases}$
5. $\begin{cases} \sin x + 2y = 2; \\ \cos(y-1) + x = 0{,}7. \end{cases}$	6. $\begin{cases} \sin x + y = -0{,}4; \\ 2x - \cos(y+1) = 0. \end{cases}$
7. $\begin{cases} \cos x + y = 1{,}5; \\ 2x - \sin(y-0{,}5) = 1. \end{cases}$	8. $\begin{cases} \sin x - 2y = 1; \\ \cos(y+0{,}5) - x = 2. \end{cases}$
9. $\begin{cases} \sin(x+0{,}5) - y = 1; \\ \cos(y-2) + x = 0. \end{cases}$	10. $\begin{cases} \sin(y+2) - x = 1{,}5; \\ \cos(x-2) + y = 0{,}5. \end{cases}$
11. $\begin{cases} \sin y - 2x = 1{,}6; \\ \cos(x+0{,}5) + y = 0{,}8. \end{cases}$	12. $\begin{cases} \sin(x+1) - y = 1; \\ \cos y + 2x = 2. \end{cases}$
13. $\begin{cases} \sin x(x-1) = 1{,}3 - y; \\ x - \sin(y+1) = 0{,}8. \end{cases}$	14. $\begin{cases} x - \cos y = 2; \\ \cos(x-1) + y = 0{,}8. \end{cases}$

15. $\begin{cases} x+\sin y=-0{,}4; \\ 2y-\cos(x+1)=0. \end{cases}$	16. $\begin{cases} \sin x+2y=1{,}6; \\ \cos(y-1)+x=1. \end{cases}$
17. $\begin{cases} \sin y-2x=1; \\ \cos(x+0{,}5)-y=2. \end{cases}$	18. $\begin{cases} 2x-\sin(y-0{,}5)=2; \\ \cos x+y=1{,}2. \end{cases}$
19. $\begin{cases} \sin(x+2)-y=1{,}5; \\ \cos(y-2)+x=0{,}5. \end{cases}$	20. $\begin{cases} \sin(x+0{,}5)-y=1{,}2; \\ \cos(y-2)+x=0. \end{cases}$

ЗАДАНИЕ 4. АППРОКСИМАЦИЯ ФУНКЦИЙ МЕТОДОМ НАИМЕНЬШИХ КВАДРАТОВ.

Исходные теоретические сведения

Пусть требуется исследовать зависимость Y=f(X), причем величины X и Y получены экспериментально, в одних и тех же экспериментах. Обычно считают, что величины X измеряются точно, в то время как измерение величин Y содержит случайные ошибки. Это означает, что погрешность измерения X пренебрежимо мала по сравнению с погрешностью измерения Y. Задача состоит в установлении функциональной зависимости Y = f(X) по результатам измерений (Xi,Yi), где i = 1,2,...,n.

Пусть функция f(X) может быть представлена в виде $f(X,a_1,a_2,...,a_k)$, где $a_1,a_2,...,a_k$ - неизвестные параметры. Тогда результаты измерений можно представить как

$$Y_i = f(X_i,a_1,a_2,...,a_k)+\chi_i,$$

где χ_i - случайные величины, характеризующие погрешности эксперимента. Обычно предполагают, что они - независимые нормально распределенные с нулевым математическим ожиданием и одинаковыми дисперсиями.

Задача состоит в том, чтобы по опытным данным наилучшим образом определить значения параметров a_k. При этом в методе наименьших квадратов считается, что наилучшими будут те значения параметров a_k, при которых сумма квадратов отклонений расчетных величин Y от экспериментальных окажется наименьшей, т.е.

$$S = \Sigma_i\,(f(X_i,a_1,a_2,...,a_k)- Y_i)^2 \Rightarrow \min.$$

На практике метод наименьших квадратов состоит из трех этапов.

1 этап. Выдвигают гипотезу о виде функции $f(X_i,a_1,a_2,...,a_k)$. Она должна быть линейна относительно параметров. Например, функция

$$a_0+a_1X_i+a_2\,X_i^2+ ....+ a_k\,X_i^k$$

линейна по параметрам a_k, т.к. любая функция, на которую умножается любой параметр , полностью вычисляется по экспериментальным данным и не зависит от параметров.

2 этап. По имеющимся экспериментальным данным определяют численные оценки параметров. Для отыскания минимума функции S нужно приравнять нулю ее частные производные по параметрам

$$dS/da_0= 0$$

$$dS/da_1= 0$$

....

$$dS/da_k = 0$$

Если $f(X_i,a_1,a_2,...,a_k)$ линейна по параметрам, то полученная в результате дифференцирования система уравнений является системой линейных алгебраических уравнений. Решая ее методами, описанными , например, в разделе 3, можно получить значения параметров .

3 этап. Проверяют выдвинутую на первом этапе гипотезу о виде функции f, т.е. насколько эта функция адекватно описывает связь экспериментальных данных (Xi,Yi). Для этого вычисляют величину

$$F = ss/ss_{resid},$$

где ss - общая сумма квадратов, равная сумме квадратов разностей между экспериментальными значениями Y_i и средним значением Y, посчитанным по всем экспериментам, ss_{resid} - остаточная сумма квадратов, равная $\Sigma_i(f(X_i,a_1,a_2,...,a_k)- Y_i)^2$. Вычисленное значение F сравнивают с $F_{табл}$, полученным по статистическим таблицам для критерия Фишера. Если $F>F_{табл}$, то считается что вид функции был выбран правильно, т.е. математическая модель описания эксперимента адекватна.

Иногда для проверки адекватности модели используют коэффициент детерминированности r^2, который изменяется в пределах от 0 до1. Если он равен 1, то выбранная модель абсолютно адекватна экспериментальным данным; если он равен 0, то никакой связи между экспериментом и выбранной моделью нет. Обычно по степени близости r^2 к 1 судят о мере адекватности математической модели.

Пример 4.1.

Аппроксимировать полиномами первой и второй степени по методу наименьших квадратов экспериментальные данные, заданные таблицей:

x	0,1	0,2	0,3	0,4	0,5	0,6	0,7	0,8	0,9	1	1,1	1,2	1,3	1,4	1,5
y	0,3	0,5	0,82	1,43	1,49	1,85	2,01	2,6	2,72	2,85	2,89	3,11	3,18	3,21	3,23

EXCEL позволяет решить эту задачу несколькими способами. Рассмотрим вначале самый простой. Откроем новый рабочий лист и поместим заданные числовые данные в блок A3:B17, отведя столбец А под значения X, а столбец В - под значения Y. Выделим блок A2:B17 и с помощью *Мастера Диаграмм* постоим график зависимости Y от X, выбрав тип диаграммы - *XY-точечная*, формат 1. Щелкнув по кнопке *Закончить* на шаге 5 из 5, получим на рабочем листе диаграмму в виде 16 точек на плоскости X-Y. Поле диаграммы будет обведено черной рамкой. Поместим курсор мыши в виде стрелки внутрь поля диаграммы и дважды быстро щелкнем. Поле диаграммы станет активным, что подтверждается изменением рамки диаграммы - она будет заштрихована синей линией. Теперь надо расположить острие стрелки курсора на любую из 16 точек диаграммы и щелкнуть. Изменится цвет всех 16 точек. Дадим команду меню *Вставка- Линия тренда*. Откроется диалог *Линия тренда*. Выберем тип тренда - *линейный*, щелкнув на соответствующем квадратике. В этом же диалоге выберем вкладку *Параметры*. Сделаем активными два указателя - *Показывать уравнение на диаграмме* и *Показывать значение R-квадрат на диаграмме*. И, наконец, щелкнув по кнопке ОК, получим решение на диаграмме. Как видно, на ней не только появляется непрерывная прямая, соответствующая линейной функциональной зависимости, но и формула зависимости Y(X) с числовыми коэффициентами и значение коэффициента детерминированности.

	A	B
1	Метод наименьших кавадратов	
2	x	y
3	0,1	0,3
4	0,2	0,5
5	0,3	0,82
6	0,4	1,43
7	0,5	1,49
8	0,6	1,85
9	0,7	2,01
10	0,8	2,56
11	0,9	2,72
12	1	2,85
13	1,1	2,89
14	1,2	3,11
15	1,3	3,18
16	1,4	3,21
17	1,5	3,23



Для аппроксимации исходных данных полиномом второго порядка нужно вновь дать команду меню *Вставка-Линия тренда*. В открывшемся диалоге выберем тип тренда - *полиномиальный*, в поле *Степень* - занесем число 2. Щелкнув по кнопке ОК, получим новую диаграмму с новыми формулами.

	A	B
1	Метод наименьших кавадратов	
2	x	y
3	0,1	0,3
4	0,2	0,5
5	0,3	0,82
6	0,4	1,43
7	0,5	1,49
8	0,6	1,85
9	0,7	2,01
10	0,8	2,56
11	0,9	2,72
12	1	2,85
13	1,1	2,89
14	1,2	3,11
15	1,3	3,18
16	1,4	3,21
17	1,5	3,23



К сожалению, этот способ решения задачи дает слишком мало статистической информации и не позволяет статистически оценить адекватность модели.

Пример 4.2.

Решим задачу примера 4.1 с помощью встроенной в EXCEL функции ЛИНЕЙН. Для решения задачи нам понадобится столбец с числовыми значениями X^2. Поэтому скопируем значения блока A2:A17 в блок A20:A35. В ячейку B20 занесем текст X*X, в ячейку B21 формулу = A21*A21 и скопируем ее в блок B22:B35. Наконец, в блок C20:C35 скопируем содержимое блока B2:B17. Таким образом, мы подготовили таблицу из 3 столбцов, в первом из которых содержатся числовые значения X, во втором - X^2, в третьем - Y.

Функция ЛИНЕЙН имеет 4 параметра, записываемых в скобках и разделяемых точкой с запятой

=ЛИНЕЙН(P1;P2;P3;P4),

где P1 - адреса блока Y (заметьте, пожалуйста, Y , а не X !),

P2 - адреса блока X,

P3 - если вместо P3 записано ИСТИНА, то a_0 вычисляется; если записано ЛОЖЬ - то $a_0 = 0$;

P4 - если вместо P4 записано ИСТИНА, то вычисляются всевозможные статистические параметры, если записано ЛОЖЬ - то вычисляются только коэффициенты $a_к,...,a_0$. Статистические параметры помещаются в таблицу вида

ak	ak-1	……	a_1	a_0
SEk	SEk-1		SE_1	SE_0
r^2	SEy			
F	df			
SSreg	SSresid			

В этой таблице $a_к,...,a_0$ - коэффициенты математической модели,

SEk, ...,SE_0 - значения стандартных ошибок коэффициентов математической модели,

r^2 - значение коэффициента детерминированности $0 < r^2 < 1$,

SEy - значение стандартной ошибки Y,

F - расчетное значение F-критерия.

df - степени свободы,

SSreg - регрессионная сумма квадратов, равная разности между общей суммой квадратов и остаточной суммой квадратов,

SSresid- остаточная сумма квадратов.

Проведем сначала <u>расчет линейной модели</u>. Занесем ячейку F20 текст: a_1, а в ячейку G20 - текст: a_0.

В ячейку F21 введем формулу

= ЛИНЕЙН(C21:C35;A21:A35;ИСТИНА;ИСТИНА)

и выделим блок F21:G25 под итоговую статистическую таблицу. Этот блок при выделении будет покрашен в черный цвет, за исключением ячейки F21. Сделаем строку формул этой ячейки активной, при этом рядом со строкой формул появится зеленая галочка. Теперь последовательно будем нажимать клавиши клавиатуры: “Ctrl”, затем, не отпуская ее, клавишу “Shift”, затем, не отпуская уже обе, клавишу “Enter”. В результате получим заполненную таблицу статистических коэффициентов в блоке F21:G25. Обратим внимание на содержимое ячеек F23 - там хранится коэффициент детерминированности и F24 - она содержит расчетное значение F-критерия.

Проверим адекватность рассчитанной модели. Для этого надо получить табличное значение F- критерия. Воспользуемся для этого статистической функцией EXCEL, которая называется FРАСПОБР. Сделаем текущей ячейку H24. Вызовем *Мастер функций* и из группы *Статистические* выберем FРАСПОБР. Щелкнув по кнопке *Шаг*>, обратимся к диалогу, в котором в поле *Вероятность* занесем 0,01, в поле *Степени свободы 1* - занесем число 15 (общее число заданных экспериментов), в поле *Степени свободы 2* - число 13 (это-df из статистической таблицы результатов, полученных для нашей модели). Если теперь щелкнуть по кнопке *Закончить*, то в ячейке H24 получим $F_{табл}$. Поскольку оно много меньше числа в ячейке F24, то модель следует считать адекватной.

	A	B	C	D	E	F	G	H
20	x	x*x	y			a1	a0	F таблич.
21	0,1	0,01	0,3			2,221429	0,36619	
22	0,2	0,04	0,5			0,162623	0,147859	
23	0,3	0,09	0,82			0,934868	0,272121	
24	0,4	0,16	1,43			186,5945	13	3,815387
25	0,5	0,25	1,49			13,81729	0,962648	
26	0,6	0,36	1,85					
27	0,7	0,49	2,01		a2	a1	a0	
28	0,8	0,64	2,56		-1,40627	4,471461	-0,27132	
29	0,9	0,81	2,72		0,17222	0,283366	0,098527	
30	1	1	2,85		0,99007	0,110613	#Н/Д	
31	1,1	1,21	2,89		597,988	12	#Н/Д	4,009621
32	1,2	1,44	3,11		14,6331	0,146823	#Н/Д	
33	1,3	1,69	3,18					
34	1,4	1,96	3,21					
35	1,5	2,25	3,23					

Перейдем к расчету квадратичной модели. Занесем в ячейки E27,F27,G27 тексты соответственно: a_2, a_1, a_0.

В ячейку E28 введем формулу

= ЛИНЕЙН(C21:C35;A21:B35;ИСТИНА;ИСТИНА)

и выделим блок E28:G32 под итоговую статистическую таблицу. Этот блок при выделении будет покрашен в черный цвет, за исключением ячейки E28. Далее проделаем те же манипуляции, что и при расчете линейной модели. В результате получим статистическую таблицу.

Может вызвать недоумение содержание ячеек G30:G32. В них записано #Н/Д. Тем не менее, так и предусмотрено в EXCEL при правильной работе функции ЛИНЕЙН.

Выделим ячейку H31 для расчета $F_{табл}$. При вызове функции FРАСПОБР изменится только число *Степеней свободы 2*. Оно в соответствии с полученной статистической таблицей равняется 12. Сравнив F с $F_{табл}$, увидим, что и оно гораздо больше $F_{табл}$. Поэтому и квадратичная модель статистически адекватна.

Порядок выполнения задания.

1. Изучить основные методы приближенного представления функций.
2. Для заданного варианта (см. таблицу), используя табличный процессор Microsoft Excel аппроксимировать экспериментальные данные.
3. Оформить отчет о проделанной работе.

Варианты заданий.

1	y_i	79.31	57.43	60.66	92.55	90.12	71.30	70.50	91.52	68.31	58.56
	x_i	5.84	3.82	6.19	9.22	7.87	6.29	4.43	8.91	5.34	2.21
2	y_i	82.16	61.02	44.56	82.52	99.17	70.24	63.23	66.48	48.35	40.24
	x_i	0.12	-3.48	-4.45	-6.19	1.81	-3.81	0.84	-2.08	-1.28	5.44
3	y_i	113.84	119.6	106.28	120.6	107.43	114.88	115.53	117.4	120.24	118.7
	x_i	2.95	2.60	2.69	3.01	2.44	2.51	3.37	2.98	3.20	2.62
4	y_i	65.72	58.05	60.05	55.79	50.83	47.69	44.49	59.74	56.81	45.82
	x_i	5.14	5.59	4.33	4.59	4.21	3.78	4.23	5.61	4.87	3.87
5	y_i	55.65	67.68	105.20	85.02	52.76	58.86	72.19	61.09	70.44	51.67
	x_i	9.11	9.35	8.90	9.22	8.74	8.98	8.77	9.31	8.81	9.14

6	y_i	22.81	28.42	24.95	26.96	8.78	36.55	15.77	22.89	27.99	14.45
	x_i	0.06	2.36	-3.14	2.10	-4.89	0.74	-0.22	1.63	-0.13	-4.97
7	y_i	18.31	21.92	16.93	-8.23	10.90	24.18	38.45	24.11	36.62	30.42
	x_i	-1.96	-0.76	-1.06	-2.95	-4.36	0.16	-2.66	-3.14	-2.12	-0.96
8	y_i	80.93	109.1	87.80	83.95	70.99	87.36	84.71	96.63	59.70	109.9
	x_i	1.65	2.43	0.94	1.29	-0.61	-1.14	2.45	2.80	1.78	1.50
9	y_i	-19.23	-21.4	-9.90	-19.5	-0.30	-12.04	1.14	11.26	-24.64	4.90
	x_i	3.80	0.25	0.48	5.78	4.91	1.56	0.91	5.73	1.98	1.23
10	y_i	18.93	-22.1	-10.07	20.59	7.09	4.04	-20.78	-12.9	3.69	-11.4
	x_i	2.00	2.60	3.66	1.70	3.79	2.08	1.97	1.93	2.02	3.75
11	y_i	63.96	44.39	51.20	58.44	50.15	44.51	47.25	35.24	43.28	32.03
	x_i	3.05	2.20	0.65	1.65	1.92	1.95	0.89	0.75	2.79	0.44
12	y_i	11.13	3.49	8.91	14.83	1.80	13.50	3.70	-2.40	10.00	16.04
	x_i	-0.05	-0.04	-0.88	-0.32	-0.24	-1.05	0.57	0.01	0.40	0.79
13	y_i	58.46	36.05	31.17	16.17	11.16	69.23	58.08	43.13	73.24	42.86
	x_i	0.22	-3.05	-1.76	-1.25	-0.45	-0.80	-0.26	-3.07	-1.27	-3.05
14	y_i	66.58	36.05	64.63	33.19	26.70	55.31	18.70	22.95	38.24	9.18
	x_i	3.44	1.72	2.06	3.07	0.99	7.65	2.92	3.53	4.10	-0.47
15	y_i	66.58	36.05	64.63	33.19	26.70	55.31	18.70	22.95	38.24	9.18
	x_i	-0.78	-0.38	1.54	-0.93	-0.83	1.82	-2.14	0.49	1.29	-1.22
16	y_i	18.93	-22.1	-10.07	20.59	7.09	4.04	-20.78	-12.9	3.69	-11.4
	x_i	7.03	5.98	7.10	6.92	6.69	3.66	7.60	3.61	4.20	7.29
17	y_i	11.13	3.49	8.91	14.83	1.80	13.50	3.70	-2.40	10.00	16.04
	x_i	3.72	4.21	4.17	5.64	2.95	6.85	2.01	1.92	3.57	2.95
18	y_i	18.31	21.92	16.93	-8.23	10.90	24.18	38.45	24.11	36.62	30.42
	x_i	-1.41	-1.44	0.45	-0.98	0.61	0.52	-1.48	-1.09	-1.60	0.15
19	y_i	55.65	67.68	105.20	85.02	52.76	58.86	72.19	61.09	70.44	51.67
	x_i	1.52	3.24	6.63	7.15	2.96	1.73	7.44	3.70	2.00	2.63
20	y_i	-19.23	-21.4	-9.90	-19.5	-0.30	-12.04	1.14	11.26	-24.64	4.90
	x_i	-2.25	1.90	0.18	-0.50	-1.09	0.94	-0.13	-0.50	-4.83	-5.98

ЗАДАНИЕ 5.ЧИСЛЕННОЕ ИНТЕГРИРОВАНИЕ ФУНКЦИЙ

Исходные теоретические сведения.

Формулы для приближенного вычисления определенных интегралов, называемые также квадратурными формулами, применяются очень часто. Дело в том, что для большого числа элементарных функций первообразные уже не выражаются через элементарные функции, в результате чего нельзя вычислить определенный интеграл по формуле Ньютона - Лейбница.

Для численного вычисления определенного интеграла должно быть задано:

1) формула f(X) -подынтегральной функции,

2) численные значения a- нижнего и b- верхнего пределов интегрирования,

3) численное значение E - точности вычисления интеграла.

Общий подход к решению задачи будет следующим. Определенный интеграл I представляет собой площадь криволинейной трапеции, ограниченной кривой f(X), осью X и прямыми x=a и x=b. Мы будем вычислять I, разбивая интервал от a до b на множество меньших интервалов, находя приблизительно площадь каждой малой полоски и суммируя площади этих полосок.

Разобьем интервал интегрирования на n равных частей, каждая длиной h=(b-a)/n. При этом интервал от a до b будет содержать (n+1) узлов x0,x1, ..., xn. Аппроксимируя f(X) на каждом малом интервале простейшими полиномами малого порядка, можно получить различные формулы вычисления интеграла I.

Заменяя f(X) прямыми, параллельными оси X, получим различные формулы прямоугольников. Вычисляя $Y_i = f(X_i)$ во всех узлах и учитывая, что площадь прямоугольника равна произведению основания, которое у всех элементарных интервалов одинаковое и равно h, на высоту, которое есть Y_i, получаем

для метода прямоугольников с узлом слева

$$I\text{слева} \sim h*(Y_0+Y_1+Y_2+...+Y_{n-1}),$$

для метода прямоугольников с узлом справа

$$I\text{справа} \sim h*(Y_1+Y_2+...+Y_{n-1}+Y_n),$$

для метода прямоугольников с узлом в центре

$$I\text{центр} \sim h*[f(X_0+h/2),f(X_1+h/2), ... , f(X_{n-1}+h/2)].$$

Если на каждом элементарном интервале заменить f(X) на трапецию, то

для метода трапеций

$$I\text{трапеций} \sim h*(Y_0/2+Y_1+Y_2+...+Y_{n-1}+Y_n/2).$$

Наконец, заменяя трапецию на параболу, получим формулу Симпсона

для метода парабол

$$I\text{парабол} \sim (h/3)*(Y_0+4Y_1+2Y_2+4Y_3+2Y_4...+4Y_{n-1}+Y_n),$$

причем n должно быть обязательно четным.

Все эти формулы вычисляют интеграл приближенно. Для оценки погрешности, возникающей при этих расчетах, используется **правило двойного пересчета**: вычисляют интеграл по выбранной квадратурной формуле дважды - сначала с некоторым шагом h, затем с шагом h/2, т.е. удваивают число n. Обозначив результаты вычислений через I_n и I_{2n}, сравнивают их по модулю. Если $|I_n - I_{2n}| < E$, то полагают $I \sim I_{2n}$. В противном случае расчет повторяют с шагом h/4. В качестве начальной величины шага можно рекомендовать число, близкое к $E^{(1/4)}$, учитывая что E всегда меньше 1.

Пример 5.1.

Пусть f(X) = 1/(1+X), a=0, b = 0,8 и E = 0,0001. Вычислим с помощью EXCEL определенный интеграл всеми приведенными выше методами. Как видно из формул, все они предусматривают вычисление тех или иных сумм значений f(X) с последующим умножением на коэффициенты, пропорциональные h.

Откроем новый рабочий лист. Отведем столбец A под значения X. Исходя из численного значения E, выберем начальное значение hнач = 0,1. Учитывая правило двойного пересчета, положим h = 0,05 и внесем числовой ряд по X от 0 до 0,8 с шагом 0,05 в блок A3:A18, т.к. X=0,8 в формуле не участвует. Отведем столбцы B и C для решения по формуле прямоугольников с узлом слева. Это значит, что в ячейки B3 и C3 надо внести одну и ту же формулу =1/(1+A3). Далее формулу B ячейке B3 надо скопировать в блок B4:B18, т.е. в этом столбце вычисляются значения f(X) с шагом 0,05. В столбце C значения f(X) надо вычислять с шагом 0,1, т.е. формулу из ячейки C3 надо копировать в ячейки C5,C7,C9 и т.д. Проще всего это сделать так. Оставить ячейку C4 пустой и выделить блок C3:C4. Далее этот блок из двух ячеек скопировать в блок C5:C17.

Отведем строку 20 для вычисления интегралов. Занесем в ячейку B20 формулу

=СУММ(B3:B18)*0,05, где СУММ - встроенная функция EXCEL, вызываемая *Мастером Функций*. В ячейку C20 введем формулу =СУММ(C3:C17)*0,1. Числа, появившиеся в этих ячейках, достаточно сильно различаются между собой.

	A	B	C	D	E	F	G	H	I	J	K
1	ОПРЕДЕЛЕННЫЕ ИНТЕГРАЛЫ										
2	x	Прямоугол. слева		Прямоугол.справа		Метод трапеций		Прямоугол. в центре		Метод парабол	
3	0	1	1			0,5	0,5	0,97561	0,952381	1	1
4	0,05	0,952381		0,952381		0,952381		0,930233		3,8095238	
5	0,1	0,909091	0,909091	0,909091	0,909091	0,909091	0,909091	0,888889	0,869565	1,8181818	3,6363636
6	0,15	0,869565		0,869565		0,869565		0,851064		3,4782609	
7	0,2	0,833333	0,833333	0,833333	0,833333	0,833333	0,833333	0,816327	0,8	1,6666667	1,6666667
8	0,25	0,8		0,8		0,8		0,784314		3,2	
9	0,3	0,769231	0,769231	0,769231	0,769231	0,769231	0,769231	0,754717	0,740741	1,5384615	3,0769231
10	0,35	0,740741		0,740741		0,740741		0,727273		2,962963	
11	0,4	0,714286	0,714286	0,714286	0,714286	0,714286	0,714286	0,701754	0,689655	1,4285714	1,4285714
12	0,45	0,689655		0,689655		0,689655		0,677966		2,7586207	
13	0,5	0,666667	0,666667	0,666667	0,666667	0,666667	0,666667	0,655738	0,645161	1,3333333	2,6666667
14	0,55	0,645161		0,645161		0,645161		0,634921		2,5806452	
15	0,6	0,625	0,625	0,625	0,625	0,625	0,625	0,615385	0,606061	1,25	1,25
16	0,65	0,606061		0,606061		0,606061		0,597015		2,4242424	
17	0,7	0,588235	0,588235	0,588235	0,588235	0,588235	0,588235	0,57971	0,571429	1,1764706	2,3529412
18	0,75	0,571429		0,571429		0,571429		0,56338		2,2857143	
19	0,8			0,555556	0,555556	0,277778	0,277778			0,5555556	0,5555556
20	Integr	0,599042	0,610584	0,57682	0,56614	0,587931	0,588362	0,587715	0,587499	0,5877869	0,5877896

Проведем аналогичные операции в столбцах D и E для вычисления интеграла методом прямоугольников с узлом справа. Следует обратить внимание на то, что на этот раз в формулах отсутствуют значения, равные X_0.

В столбцах F и G проведем вычисления интеграла методом трапеций. Здесь следует обратить внимание на формулы в строках 3 и 19. В них подынтегральная функция должна быть поделена на 2 согласно формуле трапеций.

Порядок выполнения задания.

1. Изучить основные методы интегрирования функций.
2. Для заданного варианта (см. таблицу), используя табличный процессор Microsoft Excel вычислить определенный интеграл по методу прямоугольников, трапеций и парабол.
3. Оформить отчет о проделанной работе.

Варианты заданий.

№ вар	**Функция f(x).**	**Интервал**	***№ вар***	**Функция f(x).**	**Интервал**
1	$x-2$	[-3,6]	*11*	$(x+3)^2$	[-4,2]
2	x^2+1	[-4,5]	*12*	$e^{(x-2)}$	[0, 4]
3	$\sin x$	[-2π, 2π]	*13*	$\sin 2x$	[-π, π]
4	$\sin x+4$	[-π, 2π]	*14*	$2x^2+3x$	[-5,0]
5	x^3+4	[-3,4]	*15*	$(1-x)^3$	[-2,4]
6	$\cos(x-3)$	[-π, π]	*16*	$\cos(2x+1)$	[-2π, π]
7	$2x^2-3$	[0,6]	*17*	$2x-1$	[3,8]
8	$\frac{1}{2}x+2$	[-3,3]	*18*	$x\cdot(3x+1)$	[0,5]
9	$2-x$	[-2.5,3.5]	*19*	$x\cdot\sin x$	[-π/2,2π]
10	x^2-x	[-3,1]	*20*	$\cos^2 x$	[0, 2π]

ЗАДАНИЕ 6. РЕШЕНИЕ СИСТЕМ ОБЫКНОВЕННЫХ ДИФФЕРЕНЦИАЛЬНЫХ УРАВНЕНИЙ.

Исходные теоретические сведения

Рассмотренные в предыдущем разделе одношаговые методы могут быть использованы для решения систем обыкновенных дифференциальных уравнений или для нахождения решения одного дифференциального уравнения высокого порядка. В последнем случае введением новых переменных оно сводится к решению системы дифференциальных уравнений первого порядка. Рассмотрим последовательно решение задачи Коши и краевой задачи применительно к системам таких уравнений.

6.1. Задача Коши.

Для получения однозначного решения системы дифференциальных уравнений должны быть заданы дополнительные условия. Их должно быть задано столько, каков порядок решаемой системы. Если все эти условия задаются в одной точке, т.е. при одном значении $X=X_0$, то такая задача называется задачей Коши. Эти дополнительные условия называются начальными условиями, а X_0 - называется начальной точкой.

Покажем применение метода Рунге-Кутта четвертого порядка для решения задачи Коши системы двух уравнений вида

$$dY/dX = f_1(X,Y,Z)$$

$$dZ/dX = f_2(X,Y,Z).$$

Начальные условия зададим в виде $Y(X_0)=Y_0$ и $Z(X_0)=Z_0$.

По аналогии с разделом 8.2 запишем формулы Рунге-Кутта для приближенного решения этой системы

$$Y_{i+1}=Y_i+h\Phi_1= (K_1+2K_2+2K_3+K_4)/6,$$

$$Z_{i+1} = Z_i+h\Phi_2=(L_1+2L_2+2L_3+L_4)/6,$$

где

$K_1= f_1(X_i,Y_i,Z_i)$ и $L_1= f_2(X_i,Y_i,Z_i)$,

$K_2= f_1(X_i +h/2,\ Y_i+hK_1/2,\ Z_i+hL_1/2)$ и $L_2= f_2(X_i +h/2,\ Y_i+hK_1/2,\ Z_i+hL_1/2)$,

$K_3= f_1(X_i +h/2,\ Y_i+hK_2/2,\ Z_i+hL_2/2)$ и $L_3= f_2(X_i +h/2,\ Y_i+hK_2/2,\ Z_i+hL_2/2)$,

$K_4= f_1(X_i +h, Y_i+hK_3,\ Z_i+hL_3)$ и $L_4= f_2(X_i +h,Y_i+hK_3,\ Z_i+hL_3)$.

В этих формулах $i = 0,1,2,...,n-1$. Для оценки погрешности используется правило двойного пересчета точно так же, как и при решении одного уравнения.

К решению подобной системы уравнений можно свести решение задачи Коши для уравнения второго порядка

$$d^2Y/dX^2 = f(X,Y,dY/dX)$$

с начальными условиями $Y(X_0)=Y_0$ и $Y'(X_0)=Z_0$.

Введем новую переменную $Z(X) = dY/dX$. Тогда исследуемое уравнение заменяется следующей системой из двух уравнений

$$dZ/dX= f(X,Y,Z)$$

$$dY/dX=Z,$$

с начальными условиями $Y(X_0)=Y_0$ и $Z(X_0)=Z_0$.

Пример 6.1.

Применяя метод Рунге-Кутта, вычислить на отрезке [1;1,5] таблицу значений решения уравнения

$$Y'' + Y'/X + Y = 0$$

с начальными условиями Y(1)=0,77 и Y'(1)=-0,5, выбрав шаг 0,1 с погрешностью 0,0001.

С помощью подстановки новой переменной Z, перейдем к решению системы уравнений

$$Y' = Z$$

$$Z' = -Z/X - Y,$$

с начальными условиями Y(1)=0,77 и Z(1)=-0,5.

Откроем новый рабочий лист EXCEL и выделим в нем столбцы A, B и C под переменные X, Y и Z. В последующие столбцы будем вычислять последовательно, значения коэффициентов K_i, $Ф_1$, L_i, $Ф_2$. Пусть значения X помещаются в блок A3:A8. Тогда в ячейки B3 и C3 занесем начальные значения Y(1) и Z(1). Формулы в остальных ячейках строки 3 и 4 приведены в таблице

Ячейка	Формула
D3	=C3
E3	=-C3/A3-B3
F3	=C3+0,1*E3/2
G3	=-F3/(A3+0,1/2)-(B3+0,1*D3/2)
H3	=C3+0,1*G3/2
I3	=-H3/(A3+0,1/2)-(B3+0,1*F3/2)
J3	=C3+0,1*I3
K3	=-J3/(A3+0,1)-(B3+0,1*H3)
L3	=D3+2*F3+2*H3+J3
M3	= E3+2*G3+2*I3+K3
B4	=B3+0,1*L3/6
C4	=C3+0,1*M3/6

Формулы в остальных строках решения получаются путем копирования.

Результаты решения задачи с шагом 0,1 приведены в таблице. Для оценки погрешности полученного решения следует провести решение вновь с шагом, равным половине начального, т.е. 0,05.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Решение системы дифференциальных уравнений												
2	x	y	z	K1	L1	K2	L2	K3	L3	K4	L4	Ф1	Ф2
3	1	0,77	-0,5	-0,5	-0,27	-0,5135	-0,255952	-0,512798	-0,255946	-0,525595	-0,240907	-3,07819	-1,534704
4	1,1	0,7187	-0,525578	-0,525578	-0,240898	-0,537623	-0,224919	-0,536824	-0,225012	-0,54808	-0,208281	-3,222553	-1,349042
5	1,2	0,66499	-0,548062	-0,548062	-0,208269	-0,558476	-0,190804	-0,557603	-0,190982	-0,567161	-0,17295	-3,34738	-1,14479
6	1,3	0,6092	-0,567142	-0,567142	-0,172935	-0,575789	-0,15433	-0,574859	-0,154587	-0,582601	-0,135569	-3,451039	-0,926338
7	1,4	0,55168	-0,582581	-0,582581	-0,135551	-0,589359	-0,116097	-0,588386	-0,116429	-0,594224	-0,096693	-3,532295	-0,697297
8	1,5	0,49281	-0,594203	-0,594203	-0,096674	-0,599037	-0,076624	-0,598034	-0,077029	-0,601906	-0,056815	-3,59025	-0,460793

Порядок выполнения задания.

1. Изучить основные методы численного решения дифференциальных уравнений.
2. Для заданного варианта (см. таблицу), используя табличный процессор Microsoft

Excel решить систему дифференциальных уравнений.

3. Оформить отчет о проделанной работе.

Варианты заданий.

№ вар.	$f_1(x, y_1, y_2)$	$f_2(x, y_1, y_2)$	$y_1(a)$	$y_2(a)$	a	b
1	y_1 / y_2	$y_1^2 - y_2$	1	1	1	3
2	e^{-y_1} / x	e^{-y_2} / y	-1	5	2	4
3	$\dfrac{y_1 + x}{y_1^2 + y_2^2}$	$\dfrac{y_1 - x}{y_1^2 + y_2^2}$	1	-1	2	4
4	$e^{y_1 y_2}$	$e^{-y_1 y_2}$	0	0	0	2
5	$\cos(y_1 y_2)$	$\sin(y_1 + y_2)$	0	0	0	2
6	$e^{-(y_1 + y_2)}$	$arctg(y_1)$	1	-1	2	4
7	$x + y_1^2$	$(y_1 - y_2)^2$	0	1	-1	1
8	$\sin(xy_2)$	$x\cos(xy_1)$	0	0	0	2
9	$y_1 + y_2$	$(1 + y_1^2 + y_2^2)^{-1}$	0	0	0	4
10	$\sin(y_1 y_2)$	$\cos(xy_1 y_2)$	-1	2	0	2
11	$arctg(\dfrac{1}{1 + y_1^2} + y_2^2)$	$\sin(y_1 y_2)$	1	0	-1	1
12	$y_1^2 + y_2^2$	$y_1 y_2$	-1	1	0	4
13	$arctg(x^2 + y_2^2)$	$\sin(x + y_1)$	0,5	1,5	0	2
14	$x^2 + y_2^2$	$xy_1 y_2$	1	0	0	5
15	$x^2 y_1 + y_2$	$\cos(y_1 + xy_2)$	-1	1	0	4
16	$\sin(x^2 + y_2^2)$	$\cos(xy_1)$	0	0	0	4
17	$\sin(y_2)$	$\cos(y_1)$	0,5	-0,5	1	3
18	$2\sqrt{3x^2 + y_1^2 + y_1}$	$\sqrt{x^2 + y_1^2 + y_2}$	0,5	1,2	0	2
19	$y_2 + \sqrt{x^2 + y_1^2}$	$y_2 \cos x - \sin 2x$	0,8	3,5	2	4
20	$x\cos(y_1 + y_2)$	$\sin(y_1 - y_2)$	-0,6	2	2	5

УЧЕБНО-МЕТОДИЧЕСКОЕ ОБЕСПЕЧЕНИЕ КУРСА

Рекомендуемая (основная) литература.

1. Волков Е.А. Численные методы: Учебное пособие. М.: Наука, 1982
2. Лапчик М.П. Численные методы: Учебное пособие для вузов. – М.: Acadimia, 2004
3. Турчак Л.И. Основы численных методов: Учебное пособие – М.: Наука, 1987
4. Демидович Б.П., Марон И.А. Основы вычислительной математики. М.: Наука, 1966.
5. Самарский А.А., Гулин А.В. Численные методы. М.: Наука, 1989.

Рекомендуемая (дополнительная) литература.

1. Вержбицкий В.М. Основы численных методов – М.: Высшая школа 2002.
2. Говорухин В., Цибулин В. Компьютер в математическом исследовании. – Спб.: Питер, 2001
3. Дьяконов В.П. Компьютерная математика. Теория и практика. – М.: Нолидж, 2001
4. Каханер Д. и др. Численные методы и программное обеспечение / Д. Каханер, К. Моулер, С. Неш; пер. с англ, под ред. Х.Д. Икрамова 20-е изд. – М.: Мир, 2001
5. Вычислительная математика: Конспект лекций. ЧГУ им. И.Н. Ульянова – Чебоксары, 1996
6. Ракитин В.И., Первуш В.Е. Практическое руководство по методам вычислений с приложением программ для ПК – М. Высшая школа, 1998

Сведения об имеющейся в библиотеке ЧГУ учебной литературе по курсу на 1.09.2009 г.

№	Название	**Наличие в библ.**
1.	Волков Е.А. Численные методы: Учебное пособие. М.: Наука, 1982	Чз -5, кх – 4
2.	Каханер Д. и др. Численные методы и программное обеспечение / Д. Каханер, К. Моулер, С. Неш; пер. с англ, под ред. Х.Д. Икрамова 20-е изд. – М.: Мир, 2001	Чз -1
3.	Лабораторный практикум по курсу «Основы вычислительной математики» - М. МЗ – ПРЕСС, 2001	Чз -1
4.	Лапчик М.П. Численные методы: Учебное пособие для вузов. – М.: Acadimia, 2004	Чз -5, кх – 3
5.	Вычислительная математика: Конспект лекций. ЧГУ им. И.Н. Ульянова – Чебоксары, 1996	Чз -5, кх – 94
6.	Ракитин В.И., Первуш В.Е. Практическое руководство по методам вычислений с приложением программ для ПК – М. Высшая школа, 1998	Чз -4
7.	Турчак Л.И. Основы численных методов: Учебное пособие – М.: Наука, 1987	Чз -5